




GUIA DE INICIACION

ZX SPECTRUM+

Genara
Pérez Villota



 MicroTextos

ZX Spectrum +



• USO DEL TECLADO • PROGRAMACION
BASICA • GRAFICOS • SONIDO
• JUEGOS

ZX SPECTRUM +: GUIA DE INICIACION

Una edición de

MicroTextos s.a. de ediciones

P. Castellana, 166

28046-Madrid

Telef. 458 6591 Télex 49416

Editora

Erica Witschey

Redactora Técnica

G. Pérez Villota

© MicroTextos, S. A. de ediciones, Madrid, 1985

All rights reserved

Esta publicación no puede ser reproducida, ni en todo ni en parte, ni registrada en, o transmitida por, un sistema de recuperación de información, en ninguna forma ni por ningún medio, sea mecánico, fotoquímico, electrónico, magnético, electroóptico, por fotocopia, o cualquier otro, sin el permiso previo por escrito de la editorial.

ISBN: 84-86376-15-7

Depósito Legal: M-33276-1985

Fotocomposición


Pérez-Díaz s.a. Madrid

Impresión y Encuadernación

Unigraf, S.A., Madrid

Printed in Spain

795




Genara PEREZ VILLOTA

 **MicroTextos**

Contenido

1. Cómo empezar	4
Cómo escribir una instrucción simple; Cómo escribir una lista de instrucciones; Tres programas.	
2. Cómo trabajar con el ordenador	17
Matemáticas sencillas; Variables numéricas; Otras variables numéricas; Variables encadenadas; Uso conjunto de variables numéricas y encadenadas.	
3. Cómo obtener sonidos	35
Cómo obtener sonidos extraños; Cómo crear melodías; Cómo usar un grupo de líneas más de una vez.	
4. Cómo colorear palabras y fondos	39
Uso de diferentes letras coloreadas; Cómo hacer que las letras centelleen; Una distinta manera de colorear un texto.	
5. Cómo hacer dibujos	45
Cómo dibujar semicírculos; Cómo dibujar círculos; Cómo colorear las líneas; Cómo acortar el programa.	
6. Cómo dibujar con bloques de color	52
Cómo colorear un cuadro único; Cómo colorear una fila de cuadros; Cómo colorear varias filas de cuadros; Cómo usar más de un bloque de color en un programa.	
7. "FANTASMA"	58
Uso de variables de nombres largos por variables numéricas; Uso de RND.	
8. Cómo almacenar y cargar programas	65
9. Lista de palabras de programación	68
Índice	72



Introducción


Richard Graves y su familia decidieron, hace unos meses, comprar un microordenador. Los mayores querían usarlo en sus negocios; los pequeños, divertirse con diferentes juegos; todos, unirse a la era del ordenador.

Los Graves no se daban cuenta al principio de las dificultades que eso entrañaba. En seguida comprobaron que los libros de introducción al BASIC eran muy complicados... no sólo para los más jóvenes de la familia, sino también para los mayores. Cuando terminaron por dominar el BASIC y su microordenador, el padre, Richard Graves, y su hijo de 11 años, David Graves, decidieron preparar un libro distinto.

Para que tuvieras un medio fácil y rápido de aprender a manejar y programar el ZX Spectrum, escribieron, y MicroTextos ha publicado, la *Guía de iniciación: ZX Spectrum*.

Continuando la Serie, con la *Guía de iniciación: ZX Spectrum +* podrás escribir programas sencillos para tu ZX Spectrum +, usando variables numéricas y encadenadas, sonido, color y gráficos. Como los Graves, padre e hijo de 11 años, esperamos que te sirva para abrir nuevos horizontes.

Los editores



1

Cómo empezar

Cómo escribir una instrucción simple

El ordenador solamente puede hacer lo que le digas que haga. Esto se logra escribiendo instrucciones en el teclado. Cada cosa que teclees aparecerá en la pantalla. Cada palabra, letra o símbolo, aparecerá en un lugar del monitor en donde hay una letra parpadeante llamada *cursor*. La letra cursor podrá ser una **K**, una **L**, una **E**, una **G** o una **C**.

? Problema 1: Al conectar el ordenador no se vé todavía ningún cursor. Sólo se ven las siguientes palabras:

© 1982 Sinclair Research Ltd

cerca de la parte inferior de la pantalla.

✓ Pulsa la tecla CAPS SHIFT, que está a la izquierda y derecha del teclado. Verás el cursor **K** centelleante. Ahora intenta escribir una instrucción. "Vamos" puede ser una instrucción. Por ejemplo, teclea:

Vamos "Inténtalo tú solo"


Pero en cuanto pulses la letra V, tendrás problemas. Verás que en lugar de salir la letra V, en la pantalla habrá aparecido:

CLS

y después se verá el cursor **L** centelleante.

? Problema 2: Es fácil cometer errores cuando se tecla una instrucción.

✓ Mira la parte inferior izquierda del teclado y verás la tecla CAPS SHIFT. Ahora mira hacia la parte superior derecha y encontrarás la tecla cero. Esta está señalada como **0**. No la confundas con **O**, que es la mayúscula de la letra o. Desde ahora, para ayudarte a recordar esta



▲ Antes de teclear algo en el ordenador ZX Spectrum +, tiene que aparecer un cursor (que se enciende y se apaga) en la pantalla. El cursor puede ser una de estas cinco letras. Cada letra cursor tiene un uso diferente.

diferencia, en este libro todos los ceros aparecerán como Ø. En la parte superior izquierda puedes ver la palabra DELETE, que significa borrar o deshacerse de. Pon un dedo sobre la tecla CAPS SHIFT y mantenla apretada mientras presionas Ø. Siempre que hagas esto, borrarás otra palabra, letra o símbolo a la izquierda del cursor. Por lo tanto, si cometes un error, lo podrás borrar presionando CAPS SHIFT y Ø, o DELETE, hasta que haya desaparecido, tecleando luego la instrucción correcta. Ahora puedes borrar la palabra CLS. Pero ¿qué hacer para que el ordenador te diga, “Inténtalo tú solo”?


? Problema 3: El ordenador sólo entiende las instrucciones que están dadas en un lenguaje especial llamado BASIC.

✓ Afortunadamente, el ordenador Spectrum + te ayuda lo suficiente como para hacer este lenguaje fácil de aprender. Por ejemplo, sabe que para empezar la instrucción debe haber una *palabra clave* en BASIC. Así, pues, al principio de cada nueva instrucción te mostrará el cursor **[K]** centelleante. (Esto se llama estar en modo **[K]**). Las palabras clave sencillamente son palabras en BASIC que se usan a menudo.

Si presionas una tecla cuando el ordenador está en modo **[K]**, no se verá la letra de esa tecla en la pantalla. En su lugar podrás ver la palabra clave que está escrita en blanco en la parte baja de esa tecla. Así, cuando pulsaste V, el ordenador escribió la palabra clave CLS. Si observas todas las teclas, verás que no existe la palabra clave VAMOS. VAMOS no es una instrucción en BASIC. Pero en la tecla P está la palabra clave PRINT; tú puedes usar esta instrucción en su lugar. PRINT, en inglés, significa imprimir y es el término, en lenguaje BASIC, que más se parece a la orden VAMOS a hacer algo. Siempre que quieras que el ordenador escriba un mensaje en pantalla, presionarás la palabra clave PRINT y teclearás comillas ("). A continuación teclearás el mensaje y después otra vez comillas ("). Así pues, en este caso necesitas teclear:

PRINT "Inténtalo tú solo"

Mira el cursor **[K]** y recuerda que estás en modo **[K]** desde el principio. Ahora, presiona la tecla P para obtener



▲ *Necesitas la instrucción PRINT para que el ordenador imprima tus mensajes en la pantalla.*

la palabra clave PRINT. Si miras la pantalla, ahora podrás ver:


PRINT

y después verás un nuevo cursor centelleante. Ya no estás en modo **[K]**. ¡El ordenador te vuelve a ayudar! El sabe que una vez comenzada la instrucción con una palabra clave, no querrás usar otra de las claves que hay en la parte superior de las letras. El nuevo cursor que parpadea es una **[L]**. Ahora estás en modo **[L]**. Cuando estás en modo **[L]** puedes teclear *Letras*. Si quieres que sean minúsculas, simplemente tienes que apretar las teclas. Pero si quieres que cualquiera de estas letras sea MAYUSCULA, entonces tienes que pulsar con un dedo la tecla CAPS SHIFT al mismo tiempo que presionas la letra deseada.

(Más adelante en este libro, puede que necesites escribir un mensaje con todas las letras en MAYUSCULAS. Para evitar tener un dedo permanentemente sobre CAPS SHIFT, puedes pulsar la tecla CAPS LOCK que está en la cuarta fila del teclado, del lado izquierdo, al lado de una de las teclas CAPS SHIFT. Estás ahora en modo **[C]**, o modo *Mayúscula*. No habrá cambio al cursor **[K]**, pero el cursor **[L]** (de letras) será reemplazado por el cursor **[C]** (de mayúsculas). Cualquier letra que teclees en este modo será *Mayúscula*, y podrás quitar el dedo de la tecla CAPS SHIFT mientras tecleas el mensaje. Ahora, para volver al modo **[L]**, presiona sobre CAPS LOCK una vez más. Borra todas las letras que hayas escrito mientras estabas practicando el modo **[C]** de forma que lo único que quede en la pantalla sea la palabra clave PRINT.)

Cuando en un mensaje se necesitan espacios entre letras o palabras (recuerda que el mensaje es la parte *dentro* de las comillas), simplemente aprieta la tecla BREAK, en la parte superior derecha del teclado, o bien el espaciador —la tecla alargada— en la parte inferior central del teclado. Si el ordenador quiere tener espacios entre algunos símbolos o palabras *fuera* de las comillas, los pondrá por sí mismo.

? Problema 4: Lo siguiente que quieres escribir después de PRINT no es una letra, sino ". Encontrarás las



▲ *Si quieres escribir todo el mensaje en letras MAYUSCULAS, pulsa con un dedo la tecla CAPS LOCK.*

comillas (") en la parte inferior izquierda del teclado.

✓ Esto no presenta problema alguno. Simplemente presiona la tecla de las comillas y continúa con el mensaje. Sin embargo, como información adicional y que tendrás que utilizar más adelante, recuerda que para introducir cualquiera de las palabras o símbolos que están en el segundo lugar de la parte circular de las teclas, habrás de mantener un dedo presionando SYMBOL SHIFT (en la parte inferior izquierda y derecha del teclado) mientras pulsas la tecla deseada. Ahora teclea de nuevo:

PRINT "Inténtalo tú solo"

Recuerda que si cometes un error, ya sabes cómo borrarlo y escribir otra vez.


Cuando hayas introducido una instrucción no ocurrirá nada hasta que aprietes la tecla ENTER, que está en el extremo derecho del teclado. Presiona, pues, ENTER. Ahora ya puedes leer el mensaje cerca de la parte superior de la pantalla:

Inténtalo tú solo


Al mismo tiempo, aparecerá un mensaje en la parte inferior izquierda de la pantalla:

Ø OK, Ø:1

¡Esto quiere decir que todo está bien! Ahora no hay ningún cursor en la pantalla, pero lo puedes traer de nuevo a pantalla apretando ENTER o tecleando una nueva palabra clave. Sin embargo, si has cometido algún error sin darte cuenta y pulsas ENTER, al no tener sentido la instrucción en BASIC, tu mensaje no aparecerá en la parte superior de la pantalla. En cambio, tu instrucción se habrá quedado donde estaba y aparecerá un signo de interrogación parpadeante que indica que el ordenador ha encontrado un error. Este signo de interrogación normalmente aparecerá en el lugar donde el error se ha cometido. Si esto sucede, tienes que corregir la línea borrando el error de la forma habitual y presionando ENTER otra vez. Para ver cómo funciona, intenta teclear la siguiente instrucción:




▲ Cada vez que teclees una instrucción, tienes que presionar ENTER. Enseguida el ordenador hará lo que le has pedido y escribirá un mensaje diciéndote que la instrucción ha sido recibida.





▼ Si intentas introducir (ENTER) una instrucción con algún error, el ordenador borrará la línea y en su lugar escribirá un mensaje diciendo que algo va mal.



PRINT "Hoy es jueves

y aprieta ENTER. Como debería haber habido comillas al final de la línea, la instrucción se queda donde está y aparece un signo de interrogación centelleante después de jueves. Este error está al final de la línea, por tanto no hay que borrar nada. Agrega las comillas y el signo de interrogación desaparecerá. La línea está ya bien y se puede introducir (ENTER).

Si la línea tiene un grave error, el ordenador puede decidir que es mejor empezar la línea entera otra vez. Intenta teclear:

PRINT Estoy bien

y presiona ENTER. Esta vez olvidaste teclear los dos caracteres de comillas, y el ordenador ha suprimido la línea entera y escrito un mensaje al final de la pantalla diciendo:

2 Variable not found, Ø:1

Ahora tendrás que apretar ENTER para borrar la pantalla y empezar de nuevo. Ya has aprendido a escribir una instrucción simple. Ahora prueba alguna instrucción de tu propia cosecha, usando la instrucción PRINT. Por ejemplo, pídele al ordenador que escriba el nombre de un país que desees visitar.

Cómo escribir una lista de instrucciones

Escribir una sola instrucción es bastante aburrido. Si quieres que el ordenador haga algo más interesante, necesitas darle una lista de instrucciones.

? Problema 5: El ordenador obedece cada instrucción en cuanto presionas ENTER. ¿Cómo hacer para que espere hasta haberle dado una lista de instrucciones?

✓ Teclea un número antes de cada instrucción o línea. Así el ordenador almacenará en su memoria cada línea numerada. Empieza con 1Ø para la primera línea, 2Ø para la segunda y así sucesivamente. Se pueden escribir números si estás en modo **[K]** o **[L]** pulsando las te-

clas de números en la parte superior del teclado. Si comienzas una instrucción con un número, el ordenador permanecerá en modo **[K]**. El sabe que un número al principio de una instrucción es un número de línea y continúa en espera de la palabra clave.

Introduce ahora esta lista de instrucciones, y recuerda apretar ENTER después de cada línea. Si cada línea está correcta, aparecerá en lo alto de la pantalla, donde gradualmente figurará tu lista de instrucciones. Al mismo tiempo, el mensaje 'OK' aparecerá en la parte inferior izquierda. Pulsa ENTER y empieza con la pantalla limpia.

No te preocupes si tecleas una instrucción y no cabe en una sola línea de la pantalla. Simplemente no pulses ENTER cuando no haya suficiente espacio en la línea y continúa tecleando. Algunas de las palabras saldrán en la siguiente línea de la pantalla, pero el ordenador las tratará todas como una sola línea, pues todavía no has presionado ENTER.

El . (punto) está en la cuarta fila del teclado, a la derecha, junto a CAPS SHIFT. El + (signo más), el ? (signo de interrogación), el - (guión) y el ! (signo de exclamación) están en las teclas K, C, J y I respectivamente. Recuerda usar SYMBOL SHIFT para estos últimos.

```
10 PRINT "Soy un ordenador ZX Spectrum +"
20 PRINT "Espero que te guste usarme"
30 PRINT "Sabes lo que significa el BASIC?"
40 PRINT "Beginners All-purpose"
50 PRINT "Symbolic Instruction Code!"
60 STOP
```


La palabra clave STOP de la línea 60 dice al ordenador que la lista de instrucciones ha terminado. STOP está en segundo lugar de la parte circular de la tecla A. Por tanto, debes pulsar la tecla SYMBOL SHIFT al mismo tiempo que aprietas la A.

? Problema 6: Ahora que ya has introducido una serie de instrucciones, ¿cómo hacer para que el ordenador las obedezca?

✓ Pulsa la palabra clave RUN (en la tecla R) y después ENTER. Una lista de instrucciones se llama programa. Ya has introducido y procesado (RUN) tu primer programa.

¿Qué significa BASIC?

BASIC es el nombre resultante de las iniciales en inglés de la denominación de este lenguaje: Beginners All-purpose Symbolic Instruction Code, o Código Universal de Instrucciones Simbólicas para Principiantes.



▲ Cuando hayas escrito una lista de instrucciones, pulsa RUN para que el ordenador las obedezca.


Hacia el final de la pantalla verás un mensaje que dice que el ordenador ha llegado al fin del programa en la línea 60 :

9 STOP statement, 60:1

Si quieres echar un vistazo a la lista de instrucciones otra vez, presiona la palabra clave LIST (en la tecla K) y ENTER. Para procesar otra vez el programa, pulsa otra vez RUN y luego ENTER.

? Problema 7: ¿Cómo puedes deshacerte de un programa sin desconectar el ordenador?

✓ Presiona la palabra clave NEW (en la tecla A) y aprieta ENTER. Ahora pulsa NEW y luego ENTER y escribe un programa por ti mismo, usando la instrucción PRINT. Por ejemplo, podrías escribir un programa con los nombres de monedas extranjeras, o con una lista de los países que te gustaría visitar.



▲ Si alguno de los números en el programa 'uno' está mal, no obtendrás un cuadrado completo en tu pantalla.

Tres programas que muestran lo que puede hacer el ordenador

Teclea estos programas y observa lo que ocurre. Sirven para enseñarte lo que puede hacer el ordenador. (Después de cada programa se dan indicaciones útiles para encontrar las teclas).

UNO

10	PLOT	104,48
20	DRAW	103,0
30	DRAW	0,103
40	DRAW	-103,0
50	DRAW	0, -103

Encontrarás la palabra clave PLOT en la tecla Q y la palabra clave DRAW en la tecla W. La , (coma) es una tecla en la parte inferior derecha del teclado. Para el - (signo menos) utiliza el guión en la tecla J. Ahora, procesa (RUN) el programa. Si has introducido las líneas correctamente, verás un cuadrado en la pantalla.

? Problema 8: Cuando proceses el programa, si el cuadrado no está completo, o te parece que se ve algo raro, probablemente se deba a que has tecleado mal algún número.

✓ Pulsa LIST y observa cuidadosamente las líneas. Cuando encuentres algún error, habrás de escribir la línea entera otra vez y pulsar ENTER. El ordenador automáticamente cambiará la línea errónea por la nueva que acabas de escribir. Procesa otra vez el programa y, si no hay errores, saldrá el cuadrado completo.


Ahora pulsa LIST y añade estas líneas para colorear el cuadrado de cian (tono azul):


```
60 INK 5
70 FOR x = 3 TO 15
80 FOR y = 13 TO 25
90 PRINT AT x,y;"■"
100 NEXT y
110 NEXT x
120 STOP
```

Encontrarás la palabra clave FOR en la tecla F y la palabra clave NEXT en la tecla N. La Instrucción AT está sobre la tecla I. TO está sobre la tecla F. El = (signo igual) es un símbolo en la tecla L. El ; (punto y coma) está en la parte inferior izquierda del teclado. Todas éstas teclas son fáciles de manejar. Pero, ¿qué problema encontramos con INK en la línea 60 y con el cuadro negro ■ en la línea 90?

? Problema 9: El INK de la línea 60 aparece en segundo lugar de la parte cuadrada de la tecla X. ¿Cómo la puedes introducir en un programa?

✓ Hay muchas palabras *extras* o símbolos en la parte cuadrada de las teclas. Cuando quieras pulsar una de estas palabras o símbolos, debes primero cambiar el cursor al modo **E**. Esto se hace apretando la tecla EXTEND MODE, situada en la tercera fila a la izquierda del teclado. Entonces el cursor centelleante **K** o **L** se cambiará por el cursor **E** para indicarte que estás en modo **E** (A menudo se llama modo *Extendido*). Ahora que estás en modo **E**, si pulsas una tecla podrás escribir la primera palabra o símbolo que está en la parte cuadrada de las teclas.





▲ Puedes usar la instrucción **INK** con los símbolos de Gráficos en las teclas del 1 al 8 para colorear algunas zonas de la pantalla.


Si estando en modo **[E]** mantienes un dedo apretando la tecla CAPS SHIFT, escribirás las palabras o símbolos que están en segundo lugar de la parte cuadrada de la tecla. Así, para escribir **INK**, debes estar en modo **[E]** y pulsar la tecla **X** mientras tienes un dedo apretando CAPS SHIFT. Solamente permanecerás en modo **[E]** para el tecleo de una letra. Después irás directamente al modo **[L]**, y si quieres volver al modo **[E]** tendrás que pulsar la tecla **EXTEND MODE** de nuevo.

[?] Problema 10: El símbolo cuadrado ■ de la línea 90 es un dibujo o símbolo de Gráficos. Está sobre la tecla 8. ¿Cómo puedes teclear ese, u otro de los símbolos de Gráficos que están sobre las teclas del 1 al 7, en un programa?

[✓] Para teclear un símbolo de Gráficos, primero debes cambiar al modo **[G]**. Esto se hace pulsando la tecla **GRAPH**. **GRAPH** se encuentra en la segunda fila del teclado, a la izquierda, junto a la tecla **DELETE**. Pulsa la tecla **GRAPH** y entrarás al modo **[G]** (de Gráficos). Tan pronto como estés en modo **[G]**, el cursor centelleante cambiará de **[K]** o **[L]** a **[G]**.

Ahora, antes de teclear uno de los símbolos de Gráficos, tendrás que decidir si quieres escribirlo exactamente como se muestra en la tecla con las zonas blancas en blanco y las negras en negro, o si quieres escribir *lo contrario*, con las zonas blancas en negro, y las negras en blanco. Para dibujar el *opuesto* a la tecla, pulsa el número donde está el símbolo deseado. Para dibujarlo como está, mantén apretado CAPS SHIFT mientras pulsas el número. Así, para escribir ■ deberás cambiar a modo **[G]** y pulsar simultáneamente CAPS SHIFT y el 8. Permanecerás en modo **[G]** hasta que le digas al ordenador que deseas salir del modo **[G]**. Para hacer esto, simplemente aprieta otra vez la tecla de Gráficos **GRAPH**. Estarás entonces en modo **[K]** o **[L]** una vez más.

Procesa ahora el programa. Verás como el cuadro se llena de color. Pulsa **NEW** y luego **ENTER** cuando quieras salir de este programa. (Probablemente encontrarás que es bastante duro empezar a trabajar con el teclado del Spectrum+. No te preocupes; con el uso llega a ser más fácil de manejar.)



DOS


```

10 LET x = INT (RND*22)
20 LET y = INT (RND*32)
30 LET z = INT (RND*7)
40 BORDER 1
50 INK z
60 PRINT AT x,y;"■"
70 PAUSE 5
80 GOTO 10
90 STOP

```

▲ Debido a la instrucción `GOTO 10`, este programa va dando vueltas sin llegar a un final, dibujando diferentes cuadrados coloreados en la pantalla.

Encontrarás la palabra clave `LET` sobre la tecla `L` y `BORDER` sobre la tecla `B`. `PAUSE` está sobre la tecla `M`. `GOTO` está sobre la `G`. `INT` y `RND` se encuentran en la parte cuadrada de las teclas `R` y `T`; necesitas entrar en modo `[E]` para teclearlas. El símbolo `*` está en la parte redonda de la tecla `B` y los `()` (abrir y cerrar paréntesis) están en la parte redonda de las teclas `8` y `9`. (Recuerda entrar en modo `[E]` y mantener un dedo presionado sobre `CAPS SHIFT` para escribir `INK`).



▲ Si quieres interrumpir un programa mientras se está procesando, pulsa esta tecla.

Este programa dibujará un sin fin de cuadros coloreados sobre fondo blanco, rodeados de un borde azul oscuro. La primera vez que lo proceses, verás que no aparecen cuadrados en un estrecho espacio al fondo del área blanca. Más adelante verás una razón de esto. El programa sigue y sigue porque cuando llega a la línea 80 el ordenador lee la instrucción GOTO 10. Esto le dice que salte de inmediato a la línea 10 y así va dando vueltas y vueltas. Nunca llega a la línea 90. Esto se conoce como un *bucle*.

? Problema 11: ¿Cómo se puede parar un programa que se está procesando sin tener que desconectar el ordenador?

✓ No se necesita nada más que apretar la tecla BREAK. El programa dejará de procesarse y aparecerá un mensaje en el espacio vacío, en la parte inferior de la pantalla, similar a:

L BREAK into program, 20:1

El número 20 del mensaje es el número de la línea que el ordenador había leído y obedecido justo antes de que interrumpieses el programa. Podría haber sido cualquier otra línea. Ahora ya puedes procesar el programa otra vez y aparecerá gradualmente otro modelo. También podrás continuar con el mismo modelo. Para esto último, teclea la palabra CONTINUE (CONT sobre la tecla C) y pulsa ENTER. El ordenador proseguirá entonces, empezando en la línea posterior a la que mencionó en el mensaje BREAK.

Observarás que tan pronto como interrumpiste el programa, desapareció el espacio blanco y estrecho cercano al fondo de la pantalla; así, si procesas de nuevo el programa, o lo continúas, se ve mejor. Para librarte de este programa, aprieta BREAK; entonces pulsa NEW y luego ENTER.


TRES

10	BEEP	1,0
20	BEEP	.5,2
30	BEEP	.5,0
40	BEEP	1,5


```

50 BEEP 1,4
60 BEEP .5,2
70 BEEP .5,4
80 BEEP 1,5
90 GOTO 10
100 STOP

```



Encontrarás la palabra clave BEEP en la parte cuadrada de la tecla Z. Por tanto necesitarás entrar en modo **E** y, manteniendo un dedo sobre CAPS SHIFT, pulsar la Z. (Recuerda que el punto está en la cuarta fila del teclado.) Como podrás oír cuando proceses este programa, ¡el ordenador también puede hacer sonidos!

Resumen

Has ido descubriendo muchas cosas acerca del teclado. Ya sabes:

- 1) que empiezas en modo **K**, y que entonces puedes teclear los números en las teclas de números y las palabras clave que están en primer lugar de la parte redonda de las teclas de letras;
- 2) que después entras en modo **L**, y puedes teclear los números en las teclas de números y las letras en las teclas de letras (y si quieres teclear muchas letras en MAYUSCULAS puedes ir a modo **C**);
- 3) cómo teclear las palabras o símbolos que están en segundo lugar de la parte redonda de la tecla estando

en modo [K] o [L], pulsándolas mientras mantienes un dedo presionando SYMBOL SHIFT;

4) cómo entrar en modo [E] pulsando la tecla EXTEND MODE. Puedes entonces escribir las palabras que están en primer lugar de la parte cuadrada de las teclas, pulsando éstas; y las que están en segundo lugar, apretando las teclas mientras mantienes un dedo presionando sobre la tecla de CAPS SHIFT;

5) cómo entrar o salir del modo [G], pulsando la tecla GRAPH (segunda fila a la izquierda del teclado, junto a la tecla DELETE). Estando en modo [G], puedes escribir los símbolos de Gráficos que hay sobre las teclas de números pulsando éstas mientras presionas la tecla CAPS SHIFT. Para obtener los símbolos *opuestos*, simplemente presiona las teclas sin CAPS SHIFT;


6) cómo utilizar nueve útiles instrucciones: PRINT, SPACE, STOP, RUN, LIST, NEW, GOTO, BREAK, y CONTINUE;

7) cómo introducir tus instrucciones en el ordenador pulsando la tecla ENTER;

8) cómo borrar errores, usando el Ø y la tecla CAPS SHIFT o simplemente presionando DELETE;

9) cómo escribir un programa simple, usando la instrucción PRINT, y cómo hacer que el ordenador lo procese (RUN). (Más adelante en este libro, aprenderás más sobre las instrucciones PLOT, DRAW, INK, LET, BORDER, INT, RND, PAUSE y BEEP.)

**RECUERDA
ESTOS MODOS**



K

PALABRAS CLAVE

E

+

**CAPS
SHIFT**

**PALABRAS
INFERIORES DE
LA PARTE CUADRADA
DE LAS TECLAS**

L

**LETRAS Y
NUMEROS**

G

**SIMBOLOS
GRAFICOS**

C

MAYUSCULAS

E

**PALABRAS SUPERIORES DE LA PARTE
CUADRADA DE LAS TECLAS**



Cómo trabajar con el ordenador

Matemáticas sencillas

Primero localiza los signos + (más) y - (menos). El + está en la zona redonda de la tecla K; por tanto, recuerda que para escribir + necesitarás mantener un dedo sobre SYMBOL SHIFT mientras pulsas la tecla K. El - es un símbolo en la zona redonda de la tecla J.

Aquí tenemos una suma sencilla: $3 + 4$. Si introduces esto en el ordenador y aprietas ENTER, no ocurrirá nada, excepto que aparecerá una interrogación parpadeante después del signo +. Tendrás que borrar esta línea. Pero si das la instrucción correcta, el ordenador hará la suma muy rápidamente y escribirá el resultado cerca de la parte alta de la pantalla, dando su 'OK' normal abajo. Teclea:

PRINT 3 + 4

y aprieta ENTER. La respuesta aparece al momento. Pero podrías haberlo hecho mentalmente igual de rápido. Hé aquí una operación más difícil: 4×8989898 .

[?] Problema 12: Fue fácil encontrar los signos más y menos; pero ¿dónde están los signos \times (por) y \div (dividir por)?

[✓] El signo \times (por) es *. Está en la parte redonda de la tecla B. El signo \div (dividir por) es /. Está en la parte redonda de la tecla V. Ahora teclea:

PRINT 4*8989898

y presiona ENTER. En un instante aparece la respuesta: 35959592. Efectúa estas operaciones:

5×12345678

$1471365 \div 9$

Si tienes una operación con más de una parte, como 2


BORDER
B*

* = MULTIPLICAR


CLS
V/

/ = DIVIDIR POR

▲ Para teclear estos símbolos, pulsa con el dedo SYMBOL SHIFT mientras presionas las teclas B o V.



▲ Si tienes una operación con más de una parte, el ordenador hará primero las multiplicaciones y las divisiones.



▲ Pero si colocas parte de una operación larga entre paréntesis, hará antes lo que esté dentro de los paréntesis.

$\times 5 - 4 \div 2$, el ordenador hará primero las multiplicaciones y las divisiones. Así, hará $2 \times 5 = 10$; luego $4 \div 2 = 2$ y por último $10 - 2 = 8$; el resultado es 8. Pero si colocas partes de una operación larga dentro de un paréntesis (símbolos en la parte redonda de las teclas 8 y 9), ejecutará antes las partes de dentro del paréntesis. Así en $2 \times (5 - 4) \div 2$, se hará $5 - 4 = 1$; luego $2 \times 1 = 2$; luego $2 \div 2 = 1$, así que el resultado es 1. Compruébalo escribiendo las siguientes instrucciones y pulsando ENTER después de cada una.

PRINT 2*5 - 4/2
PRINT 2*(5 - 4)/2

Para poner un punto decimal en una operación, puedes usar el punto ortográfico normal. Prueba:

PRINT 35*7.25

y pulsa ENTER. La respuesta será 253.75. Intenta por tí mismo realizar más problemas. Por ejemplo, divide el número de alumnos de tu colegio por el número de profesores para conocer cuántos alumnos hay por cada profesor.

Variables numéricas

(Hacer el mismo tipo de operación, cada vez con distintos números).

Imagina que pronto será tu santo. De regalo has pedido a todo el mundo dinero y has decidido gastar en tu colección de sellos un cuarto de lo que te den, ahorrar un cuarto y gastar la mitad en discos pop. Estas cantidades son *fijas*. ¡Pero de lo que no tienes idea es de cuánto dinero te darán! Podría ser cualquier cantidad entre 1 y 50 libras, o quizás más. Debido a que el número de libras *no es fijo*, se le llama *variable*. En el programa que sigue, puedes llamar a esta *variable numérica* con la letra **a**. (Déjala fuera de las comillas de las instrucciones PRINT.)

10 PRINT "Me dan £";a
20 PRINT "Para sellos tengo £";a/4

```
3Ø PRINT "Para ahorros tengo £";a/4
4Ø PRINT "Para discos tengo £";a/2
5Ø STOP
```

Encontrarás el símbolo £ en la parte redonda de la tecla X. Date cuenta que se necesita un punto y coma entre las comillas del mensaje PRINT y la variable numérica. Podrías haber usado una coma en lugar de punto y coma, pero cuando el programa esté en marcha, la coma dejará un espacio demasiado grande entre el signo £ y la cantidad de dinero. El punto y coma le dice al ordenador que el siguiente número de la pantalla debe ponerse justo al lado de la última letra o símbolo escrito en esa línea. Procesa ahora el programa. Verás que el ordenador sólo escribirá en la pantalla:

Me dan £

y cerca de la parte inferior de la pantalla verás el siguiente mensaje:

2 Variable not found, 1Ø:1


¡Has introducido la variable **a**, pero no le has dicho al ordenador qué es **a**! Por decirlo de otra forma, no has *definido* **a**. Puedes suponer que te darán 2Ø £ por tu santo y entonces puedes añadir la línea:

7 LET a = 2Ø

Encontrarás la palabra clave LET sobre la tecla L. (Te darás cuenta ahora de por qué empiezas siempre numerando la línea con decenas. Cuando corrijas un programa, siempre habrá mucho espacio para añadir nuevas líneas.) Si procesas ahora el programa, el ordenador imprimirá:

Me dan £20
Para sellos tengo £5
Para ahorros tengo £5
Para discos tengo £10

y debajo verás el mensaje:



▲ Para hacer que el ordenador te diga cuántas libras te han dado, tienes antes que definir qué representa la variable (la letra **a**) de la línea 1Ø.

9 STOP statement, 50:1

¡Pero con este método tendrás que escribir una nueva línea 7 cada vez que supongas una nueva cantidad de dinero! Por tanto, deberías borrar la línea 7 y hacer algo más.

? Problema 13: ¿Cómo se puede borrar una línea de un programa cuando ya ha sido escrita y has pulsado ENTER?


✓ Supongamos que quieres borrar la línea 7. Si no quieres poner nada en su lugar, tecléa simplemente 7 y pulsa ENTER. Si deseas poner algo en su lugar, tecléa 7 y el resto de la nueva línea y aprieta ENTER. El ordenador inmediatamente borrará la línea 7 antigua y pondrá la nueva en su lugar. Ahora borra la línea 7 y comprueba que ha desaparecido. Introduce estas tres nuevas líneas:

```
5 PRINT "Déjame suponer el dinero de mi santo."
6 PRINT "Cuánto será?"
7 INPUT a
```

Encontrarás la palabra clave INPUT sobre la tecla I. Intenta procesar el programa. El ordenador escribirá:

Déjame suponer el dinero de mi santo.
Cuánto será?

El ordenador entonces esperará y no hará nada debido a la instrucción INPUT **a**. Está esperando que le digas el número que has decidido poner en el almacén llamado **a**. El ordenador no continuará con el programa hasta que le des una entrada. Inténtalo poniendo 20 (simplemente tecléa 20, el cual aparecerá al final de la pantalla, y presiona ENTER). Acto seguido, el ordenador proseguirá con el programa y dará las respuestas que ya tenía antes. Ahora aprieta RUN otra vez y ENTER. En esta ocasión introduce el número 32 como entrada. Las respuestas, desde luego, serán diferentes. De hecho, puedes procesar el programa una y otra vez, suponiendo cada vez una cantidad de dinero diferente, y viendo cuánto tendrás para gastar en sellos y discos y cuánto tendrás para ahorrar.



▲ *Introduciendo una instrucción INPUT en el programa, el ordenador tiene que esperar hasta que le digas cuánto dinero crees que vas a tener. Entonces calculará cuánto tienes para gastar en cada una de las cosas.*

? Problema 14: Tu programa no tiene título.

✓ Para dar un título a un programa, introduce una línea que empiece por la palabra clave REM (sobre la tecla E), seguida de ". Para este programa el título podría ser SANTO, en cuyo caso deberías añadir:


3 REM "SANTO"

El ordenador no toma ninguna acción cuando lee la palabra REM; por tanto, las líneas REM son útiles para poner líneas y recordar qué está ocurriendo cuando se mira el programa días o meses después de haberlo escrito. Este programa ahora está completo y trabajando bien. Pero antes de teclear NEW y cambiar a otra cosa, puedes usar el programa para averiguar la respuesta de otro problema.


? Problema 15: Si quieres realizar un pequeño cambio en una línea que ya has introducido, no necesitas volver a teclear la línea entera. Supongamos, por ejemplo, que quieres ver qué ocurre si el punto y coma de la línea 10 se cambia por una coma.

✓ Escribe LIST y aprieta ENTER para ver el programa "SANTO" en la pantalla. Ahora mira la fila inferior del teclado del Spectrum. A ambos lados del espaciador (barrá larga) puedes ver unas flechas apuntando hacia la izquierda, hacia la derecha, hacia abajo y hacia arriba. Puedes usar estas flechas para moverte por la pantalla cuando estés haciendo cambios en las líneas. (Hacer cambios se llama *editar*.) Para usar las flechas simplemente debes pulsarlas para llegar al sitio deseado. Estas flechas también tienen otros usos, pero estos no serán vistos ahora.

Prueba el uso de las flechas apretando las teclas de la flecha hacia arriba o hacia abajo. Mira la pantalla para ver que ha ocurrido. Entre la cifra 3 y la palabra REM de la primera línea del programa ha aparecido un apuntador que tiene la forma >. Este apuntador se llama *cursor de programa*. Tal vez te habrás dado cuenta que cuando estás escribiendo las líneas de un programa e introduciéndolas, siempre hay un > (cursor de programa) a continuación del número de la línea que acabas de introducir.



▲ *Puedes usar las teclas de las flechas para mover el cursor de la pantalla de un lado a otro siguiendo la dirección de las flechas.*



▲ *Para hacer cambios en una línea que has introducido, primero debes hacer que aparezca una copia de esa línea en la parte inferior de la pantalla. Para lograr esto, posiciona el cursor de programa entre el número y la primera instrucción de la línea que deseas cambiar. Después pulsa EDIT.*

Cuando quieres editar una línea, lo primero que tienes que hacer es utilizar las flechas para mover el cursor de programa a esa línea. La flecha hacia abajo moverá el cursor una línea hacia abajo y la flecha hacia arriba lo moverá una línea hacia arriba. Tú quieres editar la línea 10; por tanto debes presionar cuatro veces la flecha que señala hacia abajo. Ahora el > debería estar entre el 10 y el PRINT de la cuarta línea del programa. Si has pulsado la flecha una vez de más, pulsa la flecha hacia arriba una vez para lograr llevar el cursor al lugar correcto.

Lo segundo que hay que hacer es que una copia de la línea 10 aparezca en la parte inferior de la pantalla, que es donde se pueden realizar los cambios (editar). En la tercera fila del teclado, a la izquierda, junto a la tecla de EXTEND MODE, verás la tecla EDIT. Si la pulsas, obtendrás en la parte inferior de la pantalla una copia de la línea que tiene a su lado el cursor. Ahora comprueba que el cursor está al lado de la línea 10. Pulsa la tecla EDIT una vez. Al instante aparecerá en la parte inferior de la pantalla una copia de la línea 10, diciendo:

10 [K] PRINT "Me dan £";a

Podrás ver que el cursor centelleante [K] está justo a la derecha del número 10. En el punto donde este cursor parpadeante aparece, puedes borrar de una forma normal cualquier cosa que esté a su izquierda, o añadir algo nuevo. No tiene importancia si no quieres borrar o añadir nada en el punto en donde aparece el cursor parpadeante. Puedes moverlo hacia la izquierda o hacia la derecha por medio de las flechas que aparecen en la parte inferior del teclado. Y si lo mueves hacia la derecha, cambiará, para ayudarte, del cursor [K] al [L].

Quieres cambiar el punto y coma de cerca del final de la línea 10 por una coma. Por lo tanto, debes mover el cursor centelleante hacia la derecha. Pulsa la tecla con la flecha que mira hacia ese lado hasta que en la pantalla aparezca la línea:

10 PRINT "Me dan £";[L]a

Usando ahora las teclas CAPS SHIFT y Ø o DELETE, de la forma normal, borra el punto y coma. La línea se vé de la siguiente forma:

```
1Ø PRINT "Me dan £"␣a
```

Utiliza entonces la tecla correspondiente a la coma y la línea se leerá:

```
1Ø PRINT "Me dan £",␣a
```

Ya has editado la línea. El cursor que parpadea está aún en la línea, pero no importa. Pulsa ENTER, y la nueva línea 1Ø desaparecerá de la parte inferior de la pantalla y aparecerá en la lista en la parte superior, en lugar de la línea 1Ø antigua. ¡Si procesas el programa, puedes ver cómo lo has cambiado! Practica ahora variando de la misma forma las líneas 2Ø, 3Ø, y 4Ø.

Las flechas izquierda y derecha también se pueden usar para editar una línea antes de haberla introducido con ENTER. Esto puede ahorrar mucho tiempo de borrado y reescritura, especialmente si cometes un error al principio de una línea y no te das cuenta hasta que has escrito la línea entera.



▲ Debido al cambio de punto y coma por una coma, en la línea 1Ø hay mucho espacio entre el símbolo £ y el número de libras.

Otras variables numéricas


Intenta escribir un programa más complicado usando variables numéricas. Aunque normalmente veas programas escritos sin abreviar, seguidos de explicaciones, los programas no llegan de súbito a la mente totalmente desarrollados. La mayoría de los programas comienzan con unas pocas líneas y, gradualmente, la persona que los escribe añade más y más líneas hasta escribir un programa largo y complicado.

Lo primero que hay que hacer es decidir qué clase de juego quieres inventar, o qué clase de problema quieres resolver. Por ejemplo, quizás te gustaría crear un programa que te ayudase a planificar tus gastos de vacaciones. Puedes querer conocer cuánto dinero tendrás para las vacaciones y cuánto puedes gastar cada día. Para empezar, debes decidir qué información necesita ser in-

introducida en el ordenador. En este caso se necesitará saber:

- a) Cuánto dinero tienes ya.
- b) Cuánto dinero te darán como paga durante las vacaciones.
- c) Cuánto dinero quieres que te quede al final de las vacaciones.
- d) Cuántos días estarás de vacaciones.

A continuación, debes decidir cómo introducir esto al ordenador. Inténtalo con estas líneas. Recuerda copiarlas con cuidado y tecleando todas las comillas exactamente como a continuación se detalla:



```

10 PRINT "Mete el número de £"
20 PRINT "que ya tienes"
30 INPUT a
40 PRINT a
50 PRINT "Mete el dinero que te darán
como paga"
60 INPUT b
70 PRINT b
80 PRINT "Qué cantidad de dinero
quieres que"
90 PRINT "te quede al final de"
100 PRINT "las vacaciones?"
110 INPUT c
120 PRINT c
130 PRINT "Mete el número de días que"
140 PRINT "estarás de vacaciones"
150 INPUT d
160 PRINT d
    
```

Recuerda que la palabra clave INPUT está sobre la tecla I. Las líneas PRINT 40, 70, 120 y 160 no se necesitan como información de entrada, pero están allí para que salgan en la pantalla los números que elijas y así los puedas ver y recordar. Aunque aún no está terminado, procesa el programa para ver qué ocurre.

El ordenador tiene ya toda la información que necesita, pero ahora debes decirle qué debe hacer con ella. Para averiguar cuánto puedes gastar, debes indicarle que añada el dinero que tienes (a) al dinero que te darán como paga (b) y quitar entonces de $a + b$ el dinero

que quieres que te quede (c). Por tanto, aprieta LIST y añade la línea:

```
170 PRINT "Tienes £";a + b - c;" para
gastar"
```

Date cuenta que hay un punto y coma entre las segundas " (comillas) y la expresión aritmética $a + b - c$, y otro punto y coma después de $a + b - c$ y antes de las terceras ". Podrías haber usado ambas veces una coma en su lugar, pero esto dejaría un espacio demasiado grande entre el signo £ y el número de libras cuando procesaras el programa. El uso del punto y coma significa que no habrá espacios vacíos en absoluto. Sin embargo, puedes querer algún espacio entre el número de libras y la palabra *para* en la segunda parte de la instrucción PRINT. Para ello deberás pulsar una vez el espaciador justo después de escribir las terceras ". Al teclear líneas de programas, es importante copiarlas exactamente e introducir todos los espacios de forma correcta. Si te olvidas de alguno, notarás tu error en cuanto proceses el programa, y entonces tendrás que editar la línea que estaba errónea. Desde ahora, para ayudarte a escribir las líneas correctamente, el número de espacios que se deben poner en una instrucción PRINT se indicarán con un guión como este: `_`. Un `_` significa que debes de añadir un espacio.

Ahora, para averiguar cuánto te puedes gastar cada día, debes indicar al ordenador que divida la cantidad que tienes para gastar ($a + b - c$) por el número total de días que vas a estar de vacaciones (d). Así, pues, añade las líneas:

```
180 PRINT "Tendrás _£";(a + b - c)/d
190 PRINT "para _gastar _cada _día"
```

(La línea 190 no necesita un espacio entre las primeras " y la palabra *para*, ya que es una nueva línea de programa. Por tanto, en la pantalla aparecerá esta línea separada debajo de las palabras dadas en la línea 180.) El programa ya está preparado para ser procesado, pero no tiene ni principio ni final. Le puedes dar un título y un fin con las líneas:





```
5 REM "VACACIONES"
200 STOP
```

Recuerda que la palabra STOP está en la parte redonda de la tecla A. Ahora mira el programa con cuidado.

? Problema 16: ¡El programa es tan largo que no se puede ver entero en la pantalla!

✓ Pulsa LIST (sobre la tecla K) y pulsa ENTER. Verás las primeras líneas del programa hasta la línea 160. Y debajo de las líneas, en el lado izquierdo, puedes ver la palabra:

scroll?



Un scroll es, en inglés, un rollo de papel escrito. Sólo puedes ver a la vez una parte de lo escrito. Debes enrollar un extremo y desenrollar el otro para leer la parte siguiente. El ordenador te está preguntando si quieres hacer esto mismo con las líneas del programa que has almacenado en su memoria. Si pulsas cualquiera de las teclas de letras o números (con excepción de la tecla N), el ordenador 'enrollará' las líneas de la parte superior de la pantalla, y 'desenrollará' más líneas de la parte baja, de manera que puedas leer la siguiente parte del programa. Cuando no hay más programa que desenrollar o scroll, te dará uno de sus mensajes de 'OK'. Si no quieres enrollar todo el programa pero deseas ver parte de él en la pantalla para poder editarlo, pulsa la tecla BREAK. Aparecerá entonces un mensaje al final de la pantalla que dirá:

D BREAK—CONT repeats, 0:1

Podrás entonces añadir o editar las líneas que puedes ver en la pantalla.

Volvamos ahora al programa. Quizá sería una buena idea explicar lo que intenta hacer el programa a cualquier persona que lo procese. Para ello, sería útil que añadiras las líneas:

```
6 PRINT "Este _ programa _ te _ ayuda _ a _
planificar"
7 PRINT "tus_gastos_de_vacaciones"
```

Procesa ahora el programa.

? Problema 17: El programa funciona muy bien. Pero puedes mejorar su apariencia. Sería más claro si dejaras un espacio entre las dos líneas en que dices de qué se trata el programa y las líneas en que el ordenador te pide datos. También sería útil un espacio entre las líneas en que el ordenador te pide datos, y las líneas en que él da respuestas, y también entre las dos respuestas.

✓ Para dejar una línea en blanco en la pantalla, escribe PRINT sin nada detrás. Añade, por tanto:

```
8 PRINT
9 PRINT
165 PRINT
175 PRINT
```

Procesa otra vez el programa. Parece bastante mejor. ¡Pero los programadores casi nunca están satisfechos!

? Problema 18: ¿Cómo podríamos lograr que el programa fuese más útil e interesante?

✓ Usando dos instrucciones IF ... THEN. ¿Por qué? Porque, por ejemplo, si alguien da 25 como INPUT **a**, (línea 30) y 27 como INPUT **b** (línea 60), puede dar también 70 como INPUT **c** (línea 110). El ordenador entonces escribirá en la línea 170:

Tienes £—18 para gastar

Y eso sería una tontería. Usando instrucciones IF ... THEN puedes prevenir que esto ocurra. Añade estas dos líneas:

```
124 IF a + b - c <= 0 THEN PRINT
    "Es _ una _ bobada! _ No _ puedes _
    guardar _ esa _ cantidad"
125 IF a + b - c <= 0 THEN GOTO 80
```

Encontrarás la palabra clave IF en la parte redonda de la tecla U y THEN en la parte redonda de la tecla G. El símbolo <= (menor o igual que) está sobre la tecla Q. (Nunca uses por separado los símbolos < y =, o > y =,


Este programa te ayuda a planificar tus gastos de vacaciones

Mete el número de £ que ya tienes

▲ Puedes usar la instrucción PRINT para poner líneas en blanco en la pantalla. Así el programa se verá mejor cuando se procese.



▲ Para teclear estos símbolos, mientras aprietas la tecla **SYMBOL SHIFT** pulsa las teclas **Q** o **E**.



para poner \leq o \geq .) El ' (apóstrofe) está sobre la tecla del número 7.

La línea 124 dice al ordenador que si (IF) la cantidad de dinero que queda al final de las vacaciones es \leq o menos, entonces (THEN) debe dar un mensaje diciendo que se ha dado un número demasiado grande en el INPUT en c. En otras palabras, él o ella quiere tener demasiado dinero al final de las vacaciones.

La línea 125 dice al ordenador que si (IF) el dinero que queda al final de las vacaciones es \leq o menos, entonces (THEN) debe volver a la línea 80. Esto dará al que procese el programa otra posibilidad para dar un número razonable como INPUT c.

Después de introducir estas nuevas líneas, mira o lista (LIST) el programa y estúdialo para que puedas ver qué ocurre en cada línea. Después procésalo.

Variables encadenadas

(Es dar el mismo juego de instrucciones con diferentes palabras como entrada cada vez.)

Para números variables usábamos a, b, c y d, y podíamos haber usado cualquier otra letra. Para palabras variables o grupos de letras (variables encadenadas) puedes hacer lo mismo, excepto que, por ejemplo, después de la letra a debes poner el signo \$ ("cadena") que es un símbolo sobre el 4. Damos a continuación un programa que usa las variables encadenadas que puedes utilizar en cartas de agradecimiento.

(Por supuesto en la línea 270 puedes poner tu dirección y no Gran Vía 2000.)

```

10 REM "GRACIAS"
20 PRINT "Este _ programa _ te _ ayuda _
a _ escribir"
30 PRINT "una _ carta _ de _
agradecimiento."
40 PRINT
50 PRINT
60 PRINT "Introduce _ tu _ regalo. _ Era _
un"
70 INPUT a$
    
```




```

80 PRINT a$
90 PRINT "Pon _ una _ palabra _ que _
describa _ tu _ ";a$
100 INPUT b$
110 PRINT b$
120 PRINT "Quien _ te _ regaló _ el _ ";a$
130 INPUT c$
140 PRINT c$
150 PRINT "Pon _ una _ palabra _ o _
palabras _ que _ describan _ el _ tiempo."
160 INPUT d$
170 PRINT d$
180 PRINT "Pon _ una _ palabra _ o _
palabras _ que _ describan _ tu _ familia."
190 INPUT e$
200 PRINT e$
210 PRINT "Cómo _ firmarás _ tu _
nombre?"
220 INPUT f$
230 PRINT f$
240 PRINT "Qué _ fecha _ es?"
250 INPUT g$
260 CLS
270 PRINT " _ _ _ _ _ Gran _
Via _ 2000."
280 PRINT " _ _ _ _ _ ";g$
290 PRINT
300 PRINT " _ _ Querido _ ";c$
310 PRINT " _ _ _ _ _ Muchas _ gracias _
por _ el _ ";a$;". _ Es _ realmente _ ";b$;".
_ Desde _ luego _ es _ el _ mejor _ ";a$;
_ que _ he _ tenido _ jamás."
320 PRINT " _ _ _ _ _ Espero _ que _
estés _ teniendo _ buen _ tiempo. _ El _
tiempo _ aquí _ es _ ";d$;". _ Te _
interesará _ saber _ que _ en _ la _
familia _ están _ todos _ ";e$;". _ Gracias _
otra _ vez _ por _ el _ ";a$;".
330 PRINT " _ _ _ _ _ Un _ abrazo"
340 PRINT " _ _ _ _ _ de _ ";f$
350 STOP

```


Estas líneas reúnen información para la carta.




Estas líneas escriben la carta en la pantalla.

Encontrarás la instrucción CLS sobre la tecla V. Como puedes ver, hasta la línea 250 del programa se recoge in-


TV



Spectrum



Impresora



▲ *Conectando una impresora especial al ordenador, podrás imprimir sobre papel lo que hayas escrito en la pantalla.*

formación y se muestra en pantalla. Las líneas 270 hasta el final imprimen la carta. En la línea 260 puedes ver la útil instrucción CLS. Esta instrucción le dice al ordenador que limpie la pantalla. De otra forma la carta aparecería al final de la pantalla junto con las preguntas y las respuestas.

Si tuvieses una impresora, podrías procesar el programa y luego imprimir la carta en una hoja de papel; procesarlo después otra vez, dando nuevas variables encadenadas, y escribir la siguiente carta imprimiéndola en una nueva hoja de papel. Aunque no tengas impresora, puedes disfrutar procesando el programa con variables diferentes que pueden ser o no razonables. El ordenador escribirá lo que le digas que escriba. Intenta procesar el programa varias veces y haz los cambios que creas que lo pueden mejorar.

Por ejemplo, en la línea 90 dejaste dos espacios después de la palabra *tu*, de manera que al añadir el nombre del regalo, no se partiese apareciendo la primera letra al final de la línea y el resto en la siguiente línea. Sin embargo, cuando aparecen las líneas 150 y 180 en pantalla, la palabra *describan* sí está partida. Podrías añadir espacios o bien dividir cada una de estas líneas en dos líneas PRINT del programa. ¿Puedes ver otros lugares donde puedas hacer cambios de este tipo? En caso de ser así, intenta modificar el programa.

Además, la carta es bastante escueta. Quizás te hubiese gustado hacerla más larga e incluir un párrafo mencionando un libro que hayas leído o un programa de televisión que hayas visto, y decir qué te parecen. Para hacerlo debes buscar las frases adecuadas, tales como 'He visto un programa de televisión llamado Empieza y Lo encuentro extremadamente.... ¿Me pregunto qué piensas de él?' Luego añade las líneas INPUT que sean necesarias y una o varias líneas PRINT para las frases adicionales de la carta.


Uso conjunto de las variables numéricas y encadenadas

He aquí un juego que utiliza variables numéricas y encadenadas. Acuérdate de poner todas las comillas y puntos y comas.



```

10 REM "ADIVINAME"
20 PRINT "Juego _ de _ adivinanzas _ para _
DOS _ jugadores."
30 PRINT
40 PRINT "Por _ favor _ escribe _ el _
nombre _ del _ PRIMER _ jugador."
50 INPUT a$
60 PRINT a$
70 PRINT "Por _ favor _ escribe _ el _
nombre _ del _ SEGUNDO _ jugador."
80 INPUT b$
90 PRINT b$
100 PRINT
110 PRINT "Gracias, _ ";a$;" _ y _ ";b$;"."
120 PRINT b$;" _ por _ favor _ no _ mires _
a _ la _ pantalla."
130 PRINT a$;" _ por _ favor _ escribe _ un _
número _ entre _ el _ 1 _ y _ el _ 20 _ y _ luego _
pula _ ENTER."
140 INPUT a
150 PRINT "Correcto! _ Para _ continuar _
pula _ ENTER _ otra _ vez."
160 INPUT z$
170 CLS
180 PRINT "Ahora _ que _ ";b$;" _ mire _
a _ la _ pantalla."
190 REM Bucle_hasta_que_se_acierte
200 PRINT b$;" _ adivina _ cual _ es _
el _ número _ entre _ el _ 1 _ y _ el _ 20."
210 INPUT b
220 CLS
230 IF b < > a THEN PRINT "Mala _
suerte! _ ";b;" _ no _ es _ correcto."
240 IF b < > a THEN GOTO 190
250 PRINT "BIEN _ HECHO _ ";b$;" _
ACERTASTE! _ El _ número _ correcto _ era _ "a
260 PRINT
270 PRINT
280 PRINT "Quiere _ probar _ alguien _
más? _ Escribe _ y _ por _ si _ o _ n _ por _ no."
290 INPUT c$
300 IF c$="y" THEN CLS
310 IF c$="y" THEN GOTO 200
320 IF c$<>"n" THEN GOTO 280

```



330 CLS
340 STOP



▲ Si la respuesta del jugador es mayor o menor que la correcta, el ordenador escribirá el mensaje 'Mala suerte' y ofrecerá al jugador otra oportunidad.

Verás que \diamond es un símbolo en la parte redonda de la tecla W. La primera parte de este programa es muy fácil de seguir.

La línea 20 explica que es un juego de adivinanzas. Las líneas 40 a 80 dan información al ordenador sobre los nombres de los jugadores usando variables encadenadas a\$ y b\$.

En la línea 140, el primer jugador introduce un número usando la variable numérica a. (Esto almacena el número en la memoria del ordenador, en un lugar denominado 'a').

En la línea 210, el segundo jugador introduce el número que supone que es correcto, con la variable b. (Esto es almacenado en un lugar llamado 'b').

Ahora el programa no es tan fácil de seguir. Si el segundo jugador acertó, el número almacenado en b será el mismo que el almacenado en a. Pero el segundo jugador pudo haber errado. Si ha sido así, b será menor o mayor que a. El signo < quiere decir *es menor que*, y > significa *es mayor que*, y puedes utilizar <> para decir *es menor o mayor que*. Nunca utilices separados los símbolos <y> para hacer <>.


Así pues:

En la línea 230 se dice al ordenador que si (IF) el segundo jugador no adivinó el número (IF b <> a) entonces (THEN) debe imprimir (PRINT) el mensaje 'Mala suerte' en la pantalla.

En la línea 240 se dice al ordenador que si (IF) el segundo jugador no acertó (IF b <> a) entonces (THEN) debe volver a la línea 190 para dar al segundo jugador otra oportunidad para adivinar el número. Por tanto, las líneas 190 a 240 forman lo que se llama un bucle, y el ordenador dará vueltas y vueltas en círculo hasta que se dé la respuesta correcta en el INPUT b.

Hiciste que ese bucle fuese fácil de entender usando la línea REM en la línea 190 y explicando que la siguiente sección del programa es un 'bucle hasta que se acierte'. Una línea REM, que se usa dentro de un programa para ayudarte a REMemorar que parte del programa está haciendo el qué, y no como un título de programa, no necesita comillas, puede tener espacios entre palabras y puede ser tan larga como quieras.

Cuando se adivina el número correcto, esto es **b = a**, el ordenador se saltará las líneas 230 y 240 e irá directo a la línea 250, donde se le dice que escriba el mensaje 'Bien hecho'. Cuando se acierta al número, el ordenador va a la línea 280. Llega luego a más sentencias IF ... THEN. Si (IF) alguien escribe **y**, entonces (THEN) el ordenador limpiará (CLEAR) la pantalla e irá (GOTO) a la línea 20 para comenzar un nuevo juego.



? Problema 19: Pero alguien puede introducir otra letra.

✓ Si la letra es **n**, el ordenador ignorará las líneas 300, 310 y 320 y seguirá hasta la 330, la pantalla se limpiará y el programa parará (STOP). Si la letra es otra, porque por error el jugador, o bien no había pulsado **y** o **n**, o había tecleado **Y** o **N** en mayúsculas, el ordenador ignorará las líneas 300 y 310. Pero en la 320 volverá a la línea 280 para dar al jugador otra oportunidad de teclear **y** o **n**.


? Problema 20: ¿Cómo puedes hacer que el ordenador no procese un programa demasiado rápido?

✓ Las líneas 150 y 160 tratan sobre este asunto. Tú puedes parar el ordenador siempre, y por el tiempo que desees, por medio de una línea INPUT. Cuando haya una línea INPUT, el ordenador tiene que esperar hasta que se le dé una entrada. El pulsar ENTER es una forma de ello. Así, en la línea 150 le dijiste al jugador que debía pulsar ENTER si quería continuar con el juego. El pulsar la tecla ENTER le dió al ordenador la entrada que estaba esperando en la línea 160 e inmediatamente continuó con el programa.

▲ *Cuando la respuesta del jugador es correcta, el ordenador escribe un mensaje 'Bien hecho' y te ofrece jugar otra vez.*

Maneras de mejorar "ADIVINAME"

- 1) Observa si hay alguna manera de mejorar la apariencia del programa sobre la pantalla. Unas pocas líneas PRINT podrían ser útiles, por ejemplo.
- 2) Por el momento, hay algo erróneo, no solamente con la apariencia del programa, sino con la forma en que está concebido (su *estructura*). Cuando has dado una respuesta, no tienes ni idea si el número es mayor o menor que el correcto. Se haría el juego más interesante añadiendo una línea que indique al ordenador que escri-



▲ *Puedes usar instrucciones IF ... THEN para hacer que el programa dé pistas al jugador para adivinar el número correcto.*

ba un mensaje similar a: 'El número es demasiado bajo' o 'El número es demasiado alto'. Para hacerlo, puedes usar más instrucciones IF ... THEN. Inténtalo con éstas instrucciones:

```
233 IF b < a THEN PRINT "El
número _ es _ demasiado _ bajo."
235 IF b > a THEN PRINT "El
número _ es _ demasiado _ alto."
```

Los símbolos separados < (es menor que) y > (es mayor que) están en la parte redonda de las teclas R y T. (Si te gusta este juego y quieres conservarlo (SAVE) en una cinta, para poder jugar otra vez sin tenerlo que escribir, vé a la página 65).

Resumen

Ya has aprendido:

- 1) cómo usar el ordenador para matemáticas sencillas;
- 2) cómo usar variables numéricas y encadenadas;
- 3) cómo usar el punto y coma para hacer que tu programa luzca más en la pantalla cuando se procese (RUN);
- 4) cómo borrar líneas que ya han sido introducidas, y poner líneas nuevas;
- 5) cómo usar REM para dar un título al programa;
- 6) cómo editar programas usando las flechas que están en la parte inferior del teclado y con la tecla EDIT, a la izquierda en la tercera fila del teclado;
- 7) cómo enrollar un programa largo para que puedas verlo entero en la pantalla;
- 8) cómo usar PRINT para dejar líneas blancas en la pantalla;
- 9) cómo usar sentencias IF ... THEN;
- 10) cómo usar CLS para borrar la pantalla;
- 11) cómo hacer más fácil de ver lo que está ocurriendo cuando en un programa hay un bucle, incluyendo una línea REM al principio del bucle;
- 12) cómo usar la tecla ENTER como un INPUT.





Cómo obtener sonidos


¡Puedes pedirle al ordenador ZX Spectrum + que haga ruidos extraños y emita sonidos musicales! Cuando introduzcas sonido en un programa, una de las líneas podría asemejarse a ésta:

```
10 BEEP 1,0
```

10 es, por supuesto, el número de la línea.
BEEP es la clave o instrucción para el sonido.

1, (Primer número): Puedes hacer que el sonido dure más o menos tiempo usando números del 1 al 10. El primer número, 1, dice al ordenador que haga durar el sonido un segundo. El número 2 le diría que dure 2 segundos; y así sucesivamente. Incluso si no usas números decimales con frecuencia, sabrás que .5 es lo mismo que decir medio; así 1.5 quiere decir que haga durar el sonido un segundo y medio; 2.5 dos segundos y medio; y así sucesivamente. (Si estás acostumbrado a trabajar con decimales, te interesará saber que para la duración de un sonido se pueden usar todos los números entre .01 y 10.49.)

0 (Segundo número): El sonido (la nota) se elige con el segundo número, que bien puede ser de 0 a 69, o de 0 a -60. El segundo número, 0, le dice al ordenador que toque la nota 'Do central'. En la creación de sonidos musicales, 50 es el más alto que se suele usar y -18 el más bajo.



Cómo obtener sonidos extraños

Para que el ordenador haga sonidos extraños, usa notas entre 51 y 69, o entre -19 y -60. Prueba el siguiente programa:


```
10 REM "NOCHE"  
20 BEEP 1,-40  
30 BEEP 1,-30  
40 BEEP 1,67  
50 BEEP 1,-50
```

60 BEEP 1,-30
 70 BEEP 1,-40
 80 STOP

Recuerda que BEEP está en la parte cuadrada de la tecla Z, así que tendrás que ir al modo **E** y pulsar Z mientras presionas con un dedo la tecla CAPS SHIFT. Para repetir el sonido después de procesar el programa, aprieta la tecla RUN.

Cómo crear melodías

Si quieres crear melodías, tienes que usar notas entre 50 y -18. Intenta esto:



10 REM "SOLDADO"
 20 BEEP .5,7
 30 BEEP .5,4
 40 BEEP .5,0
 50 BEEP .5,9
 60 BEEP 1,7
 70 BEEP .5,5
 80 BEEP .5,2
 90 BEEP .5,-1
 100 BEEP .5,-4
 110 BEEP 1,0
 120 BEEP .5,7
 130 BEEP .5,4
 140 BEEP .5,0
 150 BEEP .5,9
 160 BEEP 1,7
 170 BEEP .5,5
 180 BEEP .5,7
 190 BEEP .5,9
 200 BEEP .5,11
 210 BEEP 1,12
 220 STOP

Cómo usar un grupo de líneas más de una vez

Puede que te hayas dado cuenta que en el programa "SOLDADO" las líneas que van de la 20 a la 70 y de la

120 a la 170 son exactamente iguales. Siempre que estés escribiendo un programa y tengas un grupo de líneas que quieras utilizar varias veces, es una buena idea escribirlas una sola vez, después de la línea STOP, poniendo una instrucción especial en el programa que las llame siempre que lo desees. La instrucción especial es GOSUB (en la tecla H) seguida del número de la primera línea del grupo de líneas que quieras llamar. Ahora escribe otra vez el programa "SOLDADO", introduciendo la instrucción GOSUB en lugar de las líneas 20 a 70 y de la 120 a 170. Así quedará:

```
10 REM "SOLDADO"
20 GOSUB 1000
30 BEEP .5,2
40 BEEP .5,-1
50 BEEP .5,-4
60 BEEP 1,0
70 GOSUB 1000
80 BEEP .5,7
90 BEEP .5,9
100 BEEP .5,11
110 BEEP 1,12
120 STOP
```

Ahora, después de STOP, deberás introducir las líneas que GOSUB llamará. Para que estas líneas resulten fáciles de ver y entender en un programa, es bueno darles un número alto que las distinga de las demás. Por eso le has dado la instrucción GOSUB 1000 en lugar de GOSUB 130. También ayudará el hacer que la primera de las líneas GOSUB sea una línea REM, lo que te hará recordar lo que pasa en las líneas siguientes. El final del grupo de líneas GOSUB siempre tiene que ser la clave RETURN, (en la tecla Y) que hará al ordenador volver otra vez al programa principal. Añade, por tanto, las siguientes líneas:

```
1000 REM Seis notas
1010 BEEP .5,7
1020 BEEP .5,4
1030 BEEP .5,0
1040 BEEP .5,9
1050 BEEP 1,7
```



▲ Cuando quieras usar un grupo de líneas varias veces en el mismo programa, las escribes como un grupo separado al final del programa. Después empleas la instrucción GOSUB seguida del número de la primera línea del grupo de líneas para que el ordenador las busque cuando las necesite.



▲ *Usando efectos de sonido puedes hacer programas mucho más interesantes. Experimenta con tantos sonidos diferentes como puedas pensar.*

```
1060 BEEP .5,5
1070 RETURN
```

En este caso solamente has ahorrado dos líneas. Pero si hubiese habido más líneas en GOSUB, o si las hubieras querido usar más de dos veces, hubieras ahorrado mucho espacio. De mayor importancia es que el empleo de GOSUB puede hacer que un programa sea mucho más fácil de seguir. Cuando un programa es fácil de seguir, se dice que está 'bien organizado' o, mejor aún, 'bien estructurado'. ¡Una buena estructura es la señal de un buen programador!

Ahora intenta componer alguna melodía hecha por tí o instruye al ordenador para que toque alguna que ya conozcas. También puedes experimentar con ruidos extraños. ¿Podrías imitar el sonido de una olla de hervir agua mientras ésta hierve? ¿O el de una puerta chirriando?

Resumen

Ya has aprendido cómo se crean melodías y ruidos extraños usando la instrucción BEEP. También has aprendido cómo hacer que un programa sea más fácil de seguir, y también algunas veces más corto, utilizando GOSUB.

4


Cómo colorear palabras y fondos


El ordenador ZX Spectrum + normalmente tiene los *textos* (palabras) en negro sobre una pantalla blanca. Pero puedes aprender cómo colorear las letras y la pantalla. La pantalla está dividida en dos zonas. El borde de fuera, alrededor de la pantalla, se llama *borde*. Cambiar el color del borde es fácil. Sólo hay que presionar la palabra clave BORDER, en la parte redonda de la tecla B, seguida de un número. Puedes saber qué número elegir ya que las teclas de números tienen los nombres de los colores sobre ellas. Puedes ver que el 1 da azul oscuro, el 2 rojo, el 3 magenta (púrpura claro), el 4 verde, el 5 cyan (azul claro), el 6 amarillo, el 7 blanco y el 0 negro. Así, pues, BORDER 2 dará un borde rojo a la pantalla y BORDER 1 dará un borde azul oscuro a la pantalla. Intenta escribir esta instrucción como una de las primeras líneas de este programa:

```
10 REM "COLOREADO"
20 BORDER 1
30 PRINT "Ahora _verás _el _borde _azul."
```


Procesa este programa y aparecerá el borde azul. La zona de dentro o *área papel* es la parte de la pantalla en que aparecerán tus programas. El color del texto en esta zona se llama *color tinta* (INK en inglés). Para cambiar el color tinta tienes que usar la palabra clave INK, que encontrarás en la parte cuadrada de la tecla X. (Vuelve al capítulo 1 si quieres refrescarte la memoria sobre cómo escribir las palabras que están en segundo lugar de la parte cuadrada de las teclas.) Después de INK usarás uno de los números de la parte de arriba del teclado, como hiciste después de la palabra clave BORDER.

Así como has cambiado el color tinta, también puedes cambiar el color papel, de forma que puedes poner la 'tinta' en rojo y el 'papel' en amarillo, por ejemplo. Para cambiar el color 'papel' debajo de cada letra escri-





▲ *El ZX Spectrum + puede colorear el borde alrededor de toda la pantalla.*



▲ *Puede imprimir números y letras coloreadas contra un fondo de distinto color.*

ta, usa la palabra clave PAPER (en la parte cuadrada de la tecla C) y después pulsa uno de los números de la parte superior del teclado. Así, INK 2 cambiará el color del texto a rojo, y PAPER 6 cambiará el fondo de cada letra escrita al amarillo. Ahora intenta añadir estas líneas a tu programa. (Verás que antes de introducir (ENTER) tus instrucciones, aparecen en blanco en la parte inferior del borde azul.)

```
40 PRINT "Pulsa _ENTER_ para _
continuar."
50 INPUT z$
60 INK 2
70 PAPER 6
80 PRINT "Ahora _puedes _ver _letras _
rojas _sobre _un _fondo _amarillo."
```

Si quieres que tu programa deje toda la zona papel en amarillo, y no sólo la parte donde haya algo escrito, entonces después de la instrucción PAPER necesitas dar la instrucción CLS. Como ya sabes, esto limpiará la pantalla. Pero hay dos cosas interesantes en esta instrucción en las que no te habrás fijado antes ya que los colores de papel y borde eran blancos. La primera es que sólo limpia la zona *papel*. La segunda es que limpia *toda* la zona papel para poder armonizar con cualquiera de los colores que se hayan escogido. Ahora intenta añadir las siguientes líneas a tu programa. (En cuanto introduzcas la primera línea, observarás que toda la zona papel sale en amarillo y todas las líneas del programa aparecen en rojo. ¡No tienes ninguna necesidad de preocuparte!) Escribe:

```
90 PRINT
100 PRINT "Pulsa _ENTER_ para _
continuar."
110 INPUT z$
120 CLS
130 PRINT "Ahora _toda _la _zona _
papel _es _amarilla."
```

Ahora, para volver otra vez a la normalidad, necesitas cambiar las zonas borde y papel a blanco y la tinta a negro. Por lo tanto, también añade estas líneas:


```

140 PRINT
150 PRINT "Pulsa _ENTER_ para _
continuar."
160 INPUT z$
170 PAPER 7
180 CLS
190 BORDER 7
200 INK 0
210 PRINT "Vuelta _a_ la _normalidad."

```

Cuando proceses tu programa, la primera parte parecerá diferente. No te preocupes; sólo procésalo otra vez. Encontrarás que la primera parte del programa es la misma que la de antes.

Uso de diferentes letras coloreadas

Así como puedes cambiar el color de todo el texto que ves en la pantalla, también puedes dar instrucciones que sólo cambien el color tinta de una, o parte de una, sentencia particular PRINT. Inténtalo añadiendo las siguientes líneas:

```


220 PRINT
230 PRINT "Pulsa _ENTER_ para _
continuar."
240 INPUT z$
250 CLS
260 PRINT INK 2; "Mi _casa _es _roja, ";
INK 1; "_pero _el _tejado _es _azul."
270 PRINT INK 4; "La _hierba _es _verde, ";
INK 3; "_no _es _magenta!!!"

```


Cuando proceses el programa, verás que después de cada instrucción INK en una línea, las palabras cambian de color. Recuerda que las partes de una sentencia PRINT se unen con comas o puntos y comas.

Cómo hacer que las letras centelleen


Para hacer que las letras centelleen usa la palabra clave FLASH (en la parte cuadrada de la tecla V), seguida del número 1. Intenta añadir estas líneas:




▲ Puedes llenar con color toda la zona central de la pantalla.



▲ Y puedes escribir las diferentes partes de una oración en más de un color.



▲ *Puedes hacer que los colores del fondo del texto centelleen.*



▲ *O puedes hacer que el color blanco más bien apagado que normalmente tiene la pantalla parpadee con un blanco brillante.*

```
280 PRINT
290 PRINT FLASH 1;"Termina_de_
decirme_cosas_que_ya_sé!"
```

O, si quieres que la línea esté en rojo y esté centelleante, cambiala a:

```
290 PRINT INK 2;FLASH 1;"Termina _
de_decirme_cosas_que_ya_sé!"
```

La instrucción FLASH puede cortarse con FLASH seguida de Ø. Intenta esto:

```
300 PRINT
310 PRINT INK 3;FLASH 1;"PARA _YA";
FLASH Ø;" _SO _IDIOTA!"
```

Y, puedes hacer el fondo más brillante usando la palabra clave BRIGHT (en la parte cuadrada de la tecla B), seguida del número 1. Añade estas líneas:

```
320 PRINT
330 PRINT INK 2;FLASH 1;BRIGHT
1;"DIJE _ QUE _ PARASES!"
```

Puesto que las palabras "DIJE QUE PARASES!" centelleen, observarás que el blanco es mucho más brillante que el blanco normal, bastante apagado, que sueles ver en la pantalla. Puedes usar BRIGHT Ø para cortar el brillo de la misma forma que usaste FLASH Ø en la línea 310.

Una distinta manera de colorear un texto

¡Es posible colorear un texto tecleando directamente en el programa las líneas en color! Para hacerlo, supongamos que en la línea:

```
340 PRINT "Yo _ he _ parado _ ya."
```

quieres que la palabra 'Yo' esté en azul marino, 'he' en rojo, 'parado' en verde y 'ya' en cian. Primero teclea:

```
340 PRINT "
```

como siempre. Después entra en modo **E** pulsando la tecla EXTEND MODE. Luego pulsa la tecla número 1 (la tecla para el azul oscuro) mientras con el dedo presionas CAPS SHIFT. El cursor centelleante **E** se habrá transformado en un cursor **L** centelleante azul. Ahora continúa con el programa tecleando:

Yo


Verás que aparece en la pantalla en azul. Ahora da un espacio. Después vete al modo **E** otra vez, y pulsa el número 2 (la tecla para el rojo) mientras mantienes un dedo en CAPS SHIFT. Enseguida tendrás un cursor **L** centelleante en rojo, y puedes escribir:

he

que aparecerá en rojo sobre la pantalla. De la misma forma, escribe 'parado' cuando tengas un nuevo cursor parpadeante **L** en verde y 'ya' al tener el cursor parpadeante **L** de color cian. Cuando llegues a las comillas al final de la sentencia PRINT, asegúrate de que aparezcan en negro en la pantalla, cambiando al modo **E** y obteniendo entonces un cursor centelleante **L** en negro, pulsando la tecla **Ø** con CAPS SHIFT. Si las dos comillas no están en negro, podrías confundirte, porque aunque lo que aparezca en la pantalla cuando proceses (RUN) el programa no cambie, podrías alterar el color de las líneas cuando des LIST al programa. Por ejemplo, ¡mira lo que pasa en LIST si cambias las comillas finales en la línea 29 \emptyset al rojo! Finalmente, para terminar este programa sobre el color, como siempre tendrás que añadir la siguiente línea:

35 \emptyset STOP

Procesa todo el programa unas cuantas veces, comparando cuidadosamente lo que ocurre en la pantalla con la lista de instrucciones de este libro. Ahora intenta experimentar con textos y fondos en color. El empleo de color hace que los programas sean más alegres y animados, así que ¿por qué no vuelves al programa "ADIVINAME" y haces que aparezcan en color algunas pa-



▲ Puedes escribir diferentes letras coloreadas directamente en las líneas del programa, cambiando el cursor **L** a un cursor **L** coloreado.


labras en la pantalla? Si conservaste la versión final en una cinta, sólo tendrás que cargarla (LOAD) en el ordenador otra vez (ve la página 67), y enseguida podrás ponerte a trabajar.

Resumen

Ya has aprendido cómo dar color a las palabras (texto) y a los fondos usando las palabras clave: INK, PAPER y BORDER.

También has aprendido cómo hacer que las letras parpadeen y que la pantalla sea más brillante usando las palabras clave FLASH y BRIGHT.

Finalmente, has aprendido la forma de aplicar color directamente a los programas, tecleando letras coloreadas entre las comillas en negro de las sentencias PRINT, dentro de las líneas de un programa.



Si quieres dibujar cosas en la pantalla, puedes usar la zona papel como tablero de dibujo. En la página 44 hay un diagrama de la zona papel de la pantalla. Piensa que esta zona es aproximadamente de 250 unidades de ancho y 170 de alto. Si quieres describir un punto en una pantalla en blanco, tienes que indicar la distancia que hay desde la parte izquierda de la zona papel hasta el punto (por ejemplo, como muestra el primer diagrama de la derecha, el punto A está a 27 unidades de la izquierda) y luego la distancia que hay desde la parte inferior de la zona papel (el punto A está a 40 unidades). Esto se llama 'dar las coordenadas'. Intenta moverte hacia el punto A, usando la instrucción PLOT, seguida de las coordenadas, con una coma entre los dos números. Teclea:

PLOT 27,40

La palabra clave PLOT está en la tecla Q. Al pulsar ENTER, un pequeño punto aparecerá en la pantalla. Ya puedes empezar a dibujar empleando la instrucción DRAW, seguida de las coordenadas de otro punto de la pantalla. Ahora que estás 'sobre' la pantalla, las coordenadas se darán desde el punto que alcanzaste. La primera de las coordenadas debe indicar la distancia a la izquierda o a la derecha, y la segunda debe dar la distancia hacia arriba o hacia abajo. En el segundo diagrama, el punto B está 14 unidades a la izquierda del punto A y a ninguna unidad hacia arriba o hacia abajo. Para ir a la izquierda o hacia abajo, usarás el signo menos. Dibuja la línea hacia el punto B.

DRAW -14,0

La palabra clave DRAW está en la tecla W. Al pulsar ENTER el ordenador dibuja una línea desde el punto donde estabas anteriormente (punto A) al nuevo punto (punto B). Ya has dibujado la primera línea de un dibujo sencillo de un coche. El dibujo de la página siguiente muestra las once primeras líneas a dibujar con las coordenadas de todos los puntos importantes desde A hasta L.





A
27,40



A
B -14,0

▲ Puedes usar los números de la rejilla para darle al ordenador las coordenadas por donde hay que dibujar las líneas. Empiezas por un punto determinado, como A (ver arriba), y después empiezas a contar desde A hasta el próximo punto (punto B).



▲ Estas son las coordenadas que se necesitan para dibujar el perfil de un coche.

Ha llegado ya el momento de empezar a convertir todo esto en un programa. Teclea estas líneas:


```

10 REM "COCHE"
20 REM El cuerpo
30 PLOT 27,40
40 DRAW -14,0
50 DRAW -10,16
60 DRAW 10,16
70 DRAW 50,0
80 DRAW 18,23
90 DRAW 75,0
100 DRAW 25,-23
110 DRAW 25,0
120 DRAW 8,-16
130 DRAW -8,-16
140 DRAW -17,0
    
```

Intenta procesar ésto de forma que puedas comprobar si tu dibujo está bien. Hasta aquí todo ha sido muy fácil. Pero ahora quieres dibujar algo más difícil. Entre los puntos L y A, quieres algo similar a lo que muestra el diagrama de abajo.

▼ Necesitarás dos semicírculos sobre las ruedas en la base del coche.

M está 37 unidades a la izquierda de L; N a 88 a la izquierda de M; A, a 37 unidades a la izquierda de N.



Cuando llegues a M, la línea recta entre M y N será bastante fácil. Pero entre L y M hay que dibujar un semicírculo, y cuando llegues a N dibujarás otro semicírculo entre N y A.

Cómo dibujar semicírculos

Para ver fácilmente cómo dibujar semicírculos, limpia la pantalla pulsando CLS y después presionando ENTER. Esto no quiere decir que vas a perder las líneas que acabas de teclear. (Se quedarán en la memoria del ordenador hasta que escribas nuevas líneas empezando por aquellos números de líneas, o bien hasta que escribas NEW y presiones ENTER, o apagues la corriente del ordenador.) Ahora, ve a un nuevo punto de la pantalla escribiendo:

PLOT 50,100

y pulsa ENTER. Supongamos que quieres dibujar un semicírculo entre este punto y un punto 150 unidades a la derecha y 0 unidades hacia arriba. Empezarás la instrucción como si fueras a trazar una línea recta entre los dos puntos (pero no pulses ENTER todavía):

DRAW 150,0

Pero luego tienes que añadir una coma y la palabra clave PI en la parte cuadrada de la tecla M. Teclea:


DRAW 150,0,PI

y pulsa ENTER. Verás que has dibujado un semicírculo entre los dos puntos, empezando por el primero, y moviéndote en el sentido *contrario a las manillas del reloj*. Si quisieras dibujar el semicírculo en el sentido *de las manillas del reloj*, tendrías que haber usado ,-PI en lugar de ,PI. Ahora puedes volver al programa. Pulsa CLS y ENTER, y después LIST y ENTER.


Para llegar de L a M necesitas un semicírculo que se mueva en sentido contrario a las agujas del reloj, hasta un punto a 37 unidades a la izquierda, y ninguna unidad hacia arriba o hacia abajo. Así, pues, teclea:



▲ Se usa la instrucción PI para dibujar un semicírculo desde un punto a otro en sentido contrario a las agujas del reloj.



▲ Cuando hayas dibujado la base del coche, el perfil será como éste.



▲ Para colocar las ventanillas tienes que usar la instrucción **PLOT** que servirá para moverte hacia una nueva posición en tu dibujo.

150 DRAW -37,0,PI

Las dos líneas siguientes van a ser fáciles:

160 DRAW -88,0

170 DRAW -37,0,PI

Has vuelto ahora al punto A, desde donde empezaste. Para dibujar las ventanillas del coche, necesitarás dibujar dos nuevas líneas, y en cada caso necesitarás cambiar a una nueva posición, usando una instrucción **PLOT**, antes de poder dibujar cada línea. Inténtalo añadiendo estas líneas:

180 REM Las ventanillas

190 PLOT 63,72

200 DRAW 118,0

210 PLOT 122,72

220 DRAW 0,23

Las líneas 190 a 200 dibujan una línea desde E a H. Puedes calcular la posición **PLOT** de E empezando con el punto A, que era 27,40, sumando los números y restando los negativos que necesitaste para llegar a E. (Así, para el primer número tienes $27 - 14 - 10 + 10 + 50$, que es igual a 63; y para el segundo número tienes $40 + 0 + 16 + 16 + 0$ que es igual a 72). Para calcular la instrucción para llegar hasta H, suma los números, restando los negativos, que necesitaste para ir de E a H. (Así, para el primer número tienes $18 + 75 + 25$ que es igual a 118; y para el segundo tienes $23 + 0 - 23$ que es igual a 0). Las líneas 210 y 220 trazan una línea desde la mitad de la línea EH hasta la mitad de la línea FG. A ver si puedes calcular por qué lo hiciste bien cuando usaste los números 122,72 y 0,23. (Sigue el mismo método que se reseñó anteriormente.) Lo siguiente que debes dibujar son las ruedas. Para ello necesitas saber como se dibujan los círculos.

Cómo dibujar círculos

Limpia la pantalla tecleando **CLS** y pulsando **ENTER**. Si quieres dibujar un círculo, primero escribe la palabra cla-

ve CIRCLE, que está en la parte cuadrada de la tecla H. Por tanto, tu instrucción empieza con:

CIRCLE

Después escribe las coordenadas de la posición del *centro* del círculo, igual que si estuvieses dando las coordenadas en una instrucción PLOT. Por tanto, si quieres que el centro del círculo esté en 80, 80, la línea deberá decir:

CIRCLE 80,80



Lo siguiente que tienes que decidir es cuál será la distancia desde el centro del círculo a su borde (el *radio*). Si quieres que el radio sea de 20, la línea dirá:

CIRCLE 80,80,20

Teclaea esto, presiona ENTER, y verás el círculo en la pantalla. Ahora puedes volver al programa. Intenta añadir estas líneas:


```
230 REM Las ruedas
240 CIRCLE 45.5,40,14
250 CIRCLE 170.5,40,14
260 STOP
```

La línea 240 dibuja la rueda izquierda. El centro del círculo se obtuvo al decidir que el centro de la rueda estuviese en la mitad de A y N. Sabes que las coordenadas de A son 27,40; y N está a 37 a la derecha de A. Necesitas, pues, añadir la mitad de 37, que es igual a 18.5, al número 27. Esto te da 45.5 para el primer número; y el segundo número será igual, ya que el punto no está ni más arriba ni más abajo que A. El tamaño del radio, 14, se obtuvo probando varios números hasta que uno de los círculos se veía correcto. La línea 250 dibuja la rueda derecha. Intenta ver por qué acertaste al usar los números 170.5,40,14. (Para hacer esto, sólo tienes que calcular las coordenadas de M como calculamos las coordenadas de A y recordar que el centro de la rueda está en medio de M y L.)



CIRCLE 80,80,20

▲ *Para dibujar un círculo, primero hay que dar las coordenadas del centro del círculo (por ejemplo, 80,80). Luego debes decirle al ordenador cómo quieres que sea de grande (qué radio quieres que tenga) el círculo (por ejemplo, 20).*



▲ Una vez que hayas completado el dibujo, puedes dibujar las líneas en color así como añadir un borde coloreado.

Cómo colorear las líneas

Ya has dibujado el coche. Pero ¿por qué no haces tu dibujo con líneas en color usando instrucciones INK? El coche podría ser rojo y las ruedas negras. Escribe:

```
25 INK 2
235 INK Ø
```

Todas las líneas después de la 25 estarán ahora dibujadas en rojo. Luego, después de la 235, estarán todas otra vez en negro. Podrías fácilmente añadir también un borde coloreado a tu dibujo. Añade:

```
24 BORDER 6
```

Cómo acortar el programa

Tienes ya un programa de 29 líneas, muchas de las cuales son extremadamente cortas. Pero puedes poner más de una instrucción en la misma línea, siempre que separes las instrucciones por : (dos puntos), que encontrarás en la parte circular de la tecla Z. Intenta hacer esto de modo que el programa todavía sea fácil de seguir. Por ejemplo, el cambio de color de las líneas es algo que puedes hacer destacar claramente, así que lo puedes dejar en su propia línea. Recuerda también que el ordenador ignora todo lo que haya en una línea después de un REM. Por tanto ¡de nada sirve tener una instrucción REM que quieras que obedezca! Puedes reacomodar el programa "COCHE" de la siguiente manera:

```
1Ø REM "EL COCHE":REM El cuerpo
2Ø BORDER 6
3Ø INK 2
4Ø PLOT 27,4Ø: DRAW -14,Ø: DRAW
-1Ø,16: DRAW 1Ø,16
5Ø DRAW 5Ø,Ø: DRAW 18,23: DRAW 75,Ø:
DRAW 25,-23
6Ø DRAW 25,Ø: DRAW 8,-16: DRAW
-8,-16: DRAW -17,Ø
7Ø DRAW -37,Ø,PI: DRAW -88,Ø: DRAW
-37,Ø,PI
```

```

80 REM Las ventanillas
90 PLOT 63,72: DRAW 118,0
100 PLOT 122,72: DRAW 0,23
110 REM Las ruedas
120 INK 0
130 CIRCLE 45.5,40,14:CIRCLE 170.5,
40,14
140 STOP

```



Ahora ya tienes un programa de sólo 14 líneas que es fácil de seguir y que ocupa menos espacio en la memoria del ordenador. Experimenta ahora con tus propios dibujos. ¿Qué tal una mariposa? ¿O una escena del espacio? ¿O un tanque? En realidad, puedes intentar dibujar cualquier cosa que te apetezca. Solo recuerda que aunque es divertido hacer pruebas, obtendrás mejores resultados si calculas por adelantado el dibujo en papel, preferiblemente milimetrado. También puedes consultar el dibujo del área de gráficos de tu pantalla en la página 44.

Resumen

Ya has aprendido cómo hacer que el ordenador dibuje líneas rectas, semicírculos y círculos usando las palabras clave: PLOT, DRAW, PI y CIRCLE.

También has aprendido a manejar los : (dos puntos) para poner más de una instrucción en una línea.


▼ *Intenta hacer dibujos por tí mismo. Recuerda que, además de usar líneas y bordes coloreados, puedes usar las instrucciones PAPER y CLS para colorear el fondo.*





Cómo dibujar con bloques de color

Filas 0 a 31
de izquierda
a derecha



Filas 0 a 21
de arriba a
abajo

▲ *Para poner bloques en la pantalla, piensa que es una fila de cuadros. Hay 32 cuadros de izquierda a derecha y 22 cuadros de arriba a abajo.*

Ya has aprendido a dibujar con líneas. Pero querrás algunas veces rellenar una zona de color. Para que esto sea fácil, el área papel de la pantalla está dividida por el ZX Spectrum + en 22 filas (desde la fila 0 arriba del todo, hasta la 21 abajo de todo). Cada fila tiene 32 cuadros (desde el 0 a la izquierda, hasta el 31 a la derecha). Ve, a la izquierda, la división de la pantalla del ZX Spectrum +. (Recuerda que en este caso, los números bajos están en la parte *alta* de la zona papel y los altos en la *baja*. Cuando dibujabas con líneas, los números más bajos estaban en la parte baja del área papel). Puedes decidir el color de la tinta a usar (lo que sea escrito) y del papel (el fondo) para cada uno de los cuadros. O, lo que es más importante, cada uno de los cuadros puede ser rellenado por cualquiera de los símbolos GRAFICOS que hay en las teclas de números.


Cómo colorear un cuadro único

Teclea:

PRINT AT 15,5; INK 3; PAPER 6; "■"

PRINT AT es la instrucción que debes usar cuando deseas escribir algo en un lugar determinado de la pantalla. Recuerda que la palabra clave AT está en la tecla I. Los números 15,5 dicen al ordenador que el cuadro en el que quieres escribir está en la fila número 15 (la decimosexta desde arriba) y es el cuadro número 5 en esa fila (el sexto cuadro desde la izquierda). Fíjate en la posición de los puntos y comas. Cuando llegues al símbolo de Gráficos ■, recuerda que necesitas entrar en modo G. Como sabes, se entra o se sale del modo G pulsando la tecla GRAPH, segunda fila a la izquierda junto a DELETE. Si después aprietas cualquier tecla entre el 1 y el 8 con CAPS SHIFT, aparecerá un símbolo de Gráficos. La zona negra que está sobre el sím-

bolo será el área rellena con el color *tinta*, y la zona blanca con el color *papel*. Si quieres que estos colores aparezcan al revés, cuando escribas el símbolo de Gráficos **no** necesitarás mantener un dedo sobre CAPS SHIFT; esto cambiará la apariencia de las áreas blancas y negras.

Recuerda poner las comillas al final de la línea para volver al modo . Luego pulsa ENTER y verás un cuadro en magenta a lo largo de la parte de arriba y amarillo en la de abajo, cerca de la esquina inferior izquierda del área papel.

Cómo colorear una fila de cuadros


Imagina que quieres llenar de color una fila de cuadros, empezando en 3,0 hasta 3,31. Las primeras líneas podrían ser estas, ¡pero no te molestes en teclearlas!

```
10 PRINT AT 3,0; INK 2; PAPER 5;"■"
20 PRINT AT 3,1; INK 2; PAPER 5;"■"
30 PRINT AT 3,2; INK 2; PAPER 5;"■"
40 PRINT AT 3,3; INK 2; PAPER 5;"■"
```


etcétera. De esta manera habría que escribir otras 28 líneas para acabar la fila. Afortunadamente, hay una manera de hacer esto más fácil y rápidamente usando el bucle FOR ... NEXT. FOR está sobre la tecla F y NEXT sobre la N. Inténtalo escribiendo estas líneas (recuerda que TO está sobre la F).

```
10 FOR y = 0 TO 31
20 PRINT AT 3, y; INK 2; PAPER 5;"■"
30 NEXT y
40 STOP
```

Cuando proceses el programa, la fila de color aparece muy deprisa. Lo que ha ocurrido es que las líneas 10 a 30 son bucles FOR ... NEXT. La primera vez que el ordenador llega a la línea 10 se le dice que a y se le han dado valores de 0 a 31. Ya que es la primera vez que alcanza la línea 10, tiene que tomar el primer valor, que es 0. Así, en la línea 20 imprime el símbolo Gráfico al cuadro 3,0. Luego, cuando llega a la 30, se le dice que vuelva a



▲ Usa los símbolos Gráficos del teclado para colorear los cuadros de tu pantalla. Recuerda que el área negra del símbolo será cualquier color tinta (INK) que elijas. El área blanca será el color papel (PAPER).



▲ Usando un bucle

FOR ... NEXT puedes hacer que el ordenador coloree una fila de cuadros sin necesidad de dar las coordenadas en cada cuadro.

buscar el siguiente (NEXT) valor de *y*. En la línea 10 descubre que el siguiente valor de *y* es 1. Así, en la línea 20 pone el símbolo Gráfico al 3,1... etc., hasta que lo haya puesto al 3,31. Luego ya no encuentra más valores de *y*; por lo que va a la línea 40 y para (STOP).

Cómo colorear varias filas de cuadros

Supongamos que quisieras colorear no sólo la fila de cuadros desde 3,0 a 3,31, sino también las cinco filas de cuadros de debajo (de 4,0 a 4,31; de 5,0 a 5,31; de 6,0 a 6,31; de 7,0 a 7,31; y de 8,0 a 8,31). Podrías hacer esto haciendo seis grupos de bucles FOR ... NEXT de tres líneas cada uno, empezando como sigue, pero no introduces estas líneas:

```
10 FOR y = 0 TO 31
20 PRINT AT 3,y; INK 2; PAPER 5; "■"
30 NEXT y
40 FOR y = 0 TO 31
50 PRINT AT 4,y; INK 2; PAPER 5; "■"
60 NEXT y
```

etcétera. De esta forma, necesitarías otras 12 instrucciones para acabar el bloque de color. Una vez más, hay un camino para hacerlo más fácil y rápidamente usando un bucle FOR ... NEXT dentro de otro FOR ... NEXT. Intentalo escribiendo estas líneas:

```
10 FOR x = 3 TO 8
20 FOR y = 0 TO 31
30 PRINT AT x,y; INK 2; PAPER 5; "■"
40 NEXT y
50 NEXT x
60 STOP
```

Cuando proceses el programa que acabas de teclear, aparecerá gradualmente el bloque de color. Lo que ha ocurrido es esto: las líneas 10 a 50 son un bucle FOR ... NEXT. La primera vez que el ordenador llega a la 10, se le dice que *x* ha tomado todos los valores de 3 a 8. Ya que es la primera vez que ha llegado a esa línea, tiene

que tomar el primer valor, que es 3. Así x ahora es igual a 3. Pero antes de que pueda proseguir con este bucle FOR ... NEXT, encuentra en la línea 20 que ha llegado a otro bucle FOR ... NEXT que va desde la línea 20 a la 40. Así, teniendo x el valor 3, tiene que seguir todo el camino de este bucle FOR ... NEXT, que le dice que coloree la línea 3. Cuando lo ha hecho, llega a la línea 50 y se le dice que vuelva para buscar el siguiente valor de x . Con un valor 4 para x , coloreará la fila 4; y así sucesivamente, hasta que haya llegado a la fila 8, cuadro 31. Entonces, al no encontrar más valores de x , irá a la línea 60 y parará (STOP).

Uso de más de un bloque de color en un programa

Imagina que quisieras usar varios bloques de color en un programa. Puedes usar el mismo juego de instrucciones para cada bloque, haciendo que algunos de los campos en esas instrucciones sean *variables*. Puedes usar este juego de instrucciones con una instrucción GO-SUB, como hiciste en el programa "SOLDADO" de la página 37. Mira otra vez las líneas para imprimir un bloque de color. Seis de los campos de estas instrucciones pueden ser representados por una variable numérica y uno por una variable encadenada. Puedes dejar que (LET):


- a = la primera fila donde los cuadros van a ser escritos;
- b = la última fila donde van a ser escritos;
- c = el primer cuadro que quieres en cada línea;
- d = el último cuadro que quieres en cada línea;
- e = el número después de INK;
- f = el número después de PAPER; y
- g\$ = el símbolo de Gráficos entre " ".

Ya puedes escribir un juego de instrucciones como éste:


```

1010 FOR x = a TO b
1020 FOR y = c TO d
1030 PRINT AT x,y; INK e; PAPER f;g$
1040 NEXT y
1050 NEXT x

```




Siempre que quieras poner un bloque de color en un programa, sencillamente necesitarás una línea para definir las variables, y luego otra línea para que te envíe un GOSUB a este juego de instrucciones. Escribe este programa que tiene siete bloques de color y dos cuadros coloreados (ve la página 53 para instrucciones sobre inversión de los símbolos de Gráficos):



```

10 REM "ASTRONAVE"
20 REM Color del casco principal
30 LET a = 8: LET b = 13: LET c = 5: LET
d = 24: LET e = 5: LET f = 3: LET g$ = "■"
40 GOSUB 1000
50 REM Estabilizadores al final de las alas
60 LET a = 1: LET b = 2: LET c = 9: LET
d = 21
70 GOSUB 1000
80 LET a = 19: LET b = 20
90 GOSUB 1000
100 REM Alas
110 LET a = 3: LET b = 7: LET c = 14: LET
d = 16: LET e = 2: LET f = 7: LET g$ = "■"
120 GOSUB 1000
130 LET a = 14: LET b = 18
140 GOSUB 1000
150 REM Sección de la cola
160 LET a = 5: LET b = 16: LET c = 25: LET
d = 27
170 GOSUB 1000
180 REM La nariz
190 LET a = 10: LET b = 11: LET c = 3: LET
d = 4
200 GOSUB 1000
210 REM Los cañones
220 PRINT AT 5,13; INK 1; "■"
230 PRINT AT 16,13; INK 1; "■"
240 STOP
1000 REM Un bloque de color
1010 FOR x = a TO b
1020 FOR y = c TO d
1030 PRINT AT x,y; INK e; PAPER f;g$
1040 NEXT y
1050 NEXT x
1060 RETURN

```



Pulsa RUN para comprobar que el programa está trabajando bien. Si no funciona adecuadamente, pulsa BREAK y pulsa LIST y comprueba cuidadosamente tu programa. Fíjate bien en todos los dos puntos y en los puntos y comas.

Ahora mira otra vez el programa. En la línea 30 tenías que definir todas las variables. Pero observarás que en la 60 sólo tenías que definir algunas de ellas. Esto es porque las otras eran iguales a como estaban en la línea 30 y no necesitaban ser definidas de nuevo. Experimenta ahora con bloques de color para hacer tus propios dibujos. ¿Qué te parece un castillo? ¿O un soldado con un casco cuadrado? Primero experimenta con bloques de color en dibujos sencillos, y luego progresa a dibujos más complicados. Recuerda que aunque es divertido improvisar, es mejor si calculas antes el dibujo en papel milimetrado.


▲ *Cuando estés planificando tus propios dibujos, es más fácil si los trazas antes, como aquí, en un papel de gráficos (milimetrado).*

Resumen

Ya has aprendido la instrucción PRINT AT y el uso de los bucles FOR ... NEXT.



“Fantasma”



Este es un largo programa para lo que básicamente es un juego de adivinanzas. Hay una casa encantada con un fantasma en una de las habitaciones y tienes que adivinar en cuál está. ¡Lo emocionante del juego consiste en que cada vez el fantasma está en una habitación diferente! También encontrarás dos nuevas formas de hacer cosas en este programa.

Uso de variables de nombres largos por variables numéricas

Hasta ahora habías usado letras sueltas en instrucciones tales como:

```
LET a = 20
```

Pero puedes usar nombres en su lugar, lo que algunas veces facilita ver lo que está ocurriendo. Por ejemplo, puedes escribir:

```
LET Altura = 5
```

o:

```
LET Fortuna = 200000
```

En “FANTASMA”, encontrarás en la línea 30 la variable *habitant* (Habitación de fantasma), y en la línea 40 la variable *Puntos*. La única regla es que las variables de nombres largos deben empezar por una letra.

Uso de RND

Esta es una instrucción muy importante que te permite inventar juegos que tengan algo de azar. Encontrarás RND sobre la tecla T. También necesitarás encontrar la palabra clave INT, que está sobre la tecla R. La siguiente

instrucción haría que el ordenador escribiese, al azar (RND), cualquier número *entero* entre 1 y 10:

```
PRINT 1 + INT (RND*10)
```

o:

```
PRINT 1 + INT (RND*8)
```

hará que el ordenador elija un número entre 1 y 8; etcétera. En el programa, hay una casa con ocho habitaciones. La habitación donde está el fantasma se llama Habifant. En la línea 30, la instrucción:

```
LET Habifant = 1 + INT (RND*8)
```


significa que el ordenador elije un número entre 1 y 8; y hasta que empiece un nuevo juego, ese número es el de la Habifant. Más adelante en el programa, en la línea 2270, el jugador tiene que adivinar ese número. Si el jugador lo adivina, el juego llega a su fin y el ordenador elije otro número entre el 1 y el 8 para la Habifant del siguiente juego.

Si no adivina, el jugador tiene otra oportunidad. Si acierta a la primera, los puntos serán 6. Cada respuesta errónea baja el resultado 1 punto. Es decir, si acierta a la segunda, tendrá 5 puntos; a la tercera, tendrá 4 puntos; etcétera. ¡Si llegas a 0 puntos, el fantasma te comerá! Por tanto, sólo tienes 6 oportunidades para adivinar el número correcto de la Habifant.

Ahora, estudia el programa cuidadosamente para calcular el orden en que suceden las cosas. Verás que se usan muchas instrucciones GOSUB, lo cual ha partido el programa en trozos, cada uno de los cuales debería ser razonablemente fácil de entender. El cuerpo principal del programa termina en la línea 130 y desde ahí todo son GOSUBs.

(Como éste es un programa muy largo, puedes querer escribirlo por etapas. Vale la pena recordar que puedes almacenar (SAVE) parte del programa y más tarde cargarlo (LOAD) de nuevo en el ordenador cuando quieras continuar. En el siguiente capítulo se explica cómo almacenar y cargar programas).

Escribe el programa que se presenta a continuación y procésalo con mucho cuidado.



▲ *La instrucción INT seguida de (RND*8) hará que el ordenador elija un número diferente cada vez que proceses el programa. El ordenador escogerá un número al azar, de forma que no conozcas cuál es.*



```

10 REM "FANTASMA"
20 GOSUB 1000
30 LET Habifant = 1 + INT (RND*8)
40 LET Puntos = 6
50 GOSUB 2000
60 IF a < > Habifant THEN GOSUB 3000
70 IF Puntos = 0 THEN GOTO 100
80 IF a < > Habifant THEN GOTO 50
90 GOSUB 4000
100 GOSUB 7000
110 IF b$ = "y" THEN GOTO 20
120 IF b$ < > "n" THEN GOTO 100
130 STOP

```

```

1000 REM Introducción
1010 BORDER 4
1020 PRINT:PRINT:PRINT
1030 PRINT INK 2; FLASH 1;"-----
-----FANTASMA-----
-----"

```

```

1040 GOSUB 6500
1050 PRINT:PRINT:PRINT
1060 PRINT INK 1;"-----
--Por"
1070 PRINT
1080 PRINT INK 1;"---La _Familia _
Graves _ (1984)"
1090 PRINT:PRINT
1100 PRINT "Cómo _ te _ llamas?"
1110 INPUT a$: PRINT a$
1120 GOSUB 6000
1130 PRINT:PRINT
1140 PRINT INK 4;"pronto, _ ;a$;" _tú _
verás"
1150 PRINT INK 4;FLASH 1;"-----
--UNA _CASA _ENCANTADA-----
--"

```

```

1160 PRINT:PRINT:PRINT
1170 GOSUB 6500
1180 PRINT INK 3;"Pulsa _ENTER _para _
continuar"
1190 INPUT z$: CLS: RETURN
2000 REM Casa fantasma y Adivinar
2010 REM La Casa

```



```

2020 LET a = 8: LET b = 19: LET c = 4: LET
d = 27: LET e = 6: LET f = 7: LET g$ = "■":
GOSUB 5000
2030 REM El tejado
2040 LET a = 4: LET b = 7: LET c = 8: LET
d = 23: LET e = 2: LET g$ = "■": GOSUB 5000
2050 LET a = 6: LET c = 6: LET d = 7:
GOSUB 5000
2060 LET c = 24: LET d = 25: GOSUB 5000
2070 PRINT AT 7,5; INK 2;"■": PRINT AT
5,7; INK 2;"■"
2080 PRINT AT 7,4; INK 2;"■": PRINT AT
6,5; INK 2;"■": PRINT AT 5,6; INK 2;"■":
PRINT AT 4,7; INK 2;"■"
2090 PRINT AT 7,26; INK 2;"■": PRINT AT
5,24; INK 2;"■"
2100 REM Ventanas de arriba
2110 LET a = 9: LET b = 12: LET c = 5: LET
d = 8: LET g$ = "■": GOSUB 5000
2120 LET c = 11: LET d = 14: GOSUB 5000
2130 LET c = 17: LET d = 20: GOSUB 5000
2140 LET c = 23: LET d = 26: GOSUB 5000
2150 REM Ventanas de abajo
2160 LET a = 15: LET b = 17: LET c = 5: LET
d = 7: GOSUB 5000
2170 LET c = 10: LET d = 12: GOSUB 5000
2180 LET c = 19: LET d = 21: GOSUB 5000
2190 LET c = 24: LET d = 26: GOSUB 5000
2200 REM La puerta
2210 LET b = 19: LET c = 14: LET d = 17: LET
e = 3: LET g$ = "■": GOSUB 5000
2220 PRINT AT 0,0; INK 4; FLASH 1;" _ _ _
_ AQUI _ ESTA _ LA _ CASA _ ENCANTADA
_ _ "
2230 PRINT INK 4;"Con _ 8 _ habitaciones. _
Dónde _ está _ el _ FANTASMA"
2240 PRINT AT 2,8; INK 1;a$, "_ ?"
2250 GOSUB 6000
2260 PRINT AT 21,0; INK 2;"AHORA _
METE _ TU _ SUPOSICION _ DESDE _ 1 _
HASTA _ 8"
2270 INPUT a: CLS : RETURN
3000 REM Puntos después de error
3010 PRINT

```

**RECUERDA
ENTRAR EN
MODO **G**
ANTES DE
ESCRIBIR ESTOS
SIMBOLOS
GRAFICOS**

**¡CONTINUA
AUN HAY MAS!**




```


3020 LET Puntos = Puntos - 1
3030 PRINT INK 2;"MALA _SUERTE _";a$
3040 PRINT:PRINT:PRINT:PRINT
3050 PRINT INK 4;"El _FANTASMA _no _
    está _en _la _habitación";FLASH 1;a
3060 PRINT:PRINT:PRINT:PRINT
3070 IF Puntos = 0 THEN PRINT FLASH 1;
    BRIGHT 1; INK 2;"AY! _TU _HAS _
    SIDO _COMIDO _POR _EL _FANTASMA"
3080 IF Puntos = 0 THEN PRINT:PRINT
3090 IF Puntos = 0 THEN GOSUB 6500
3100 IF Puntos = 0 THEN PRINT INK 1;"Por _
    favor _Pulsa _ENTER"
3110 IF Puntos = 0 THEN GOTO 3160
3120 PRINT INK 1;"Tus _posibles _puntos _
    han _bajado _de _6 _a _";INK 2; Puntos
3130 PRINT:PRINT:PRINT:PRINT
3140 GOSUB 6500
3150 PRINT INK 1;"Pulsa _ENTER _cuando _
    tú _estés _"; INK 2;"preparado _"; INK 1;" _
    para _tener _otra _oportunidad!!!"
3160 IF Puntos = 0 THEN LET a = Habifant
3170 INPUT z$: CLS : RETURN

```

```

4000 REM Puntos después de acierto
4010 PRINT PAPER 2; INK 6; FLASH 1;
    BRIGHT 1;" _ _ _ _BIEN _HECHO _ _ _ _";
    FLASH 0;" _ _ _ _";a$: GOSUB 6000
4020 PRINT
4030 PRINT INK 0; BRIGHT 1;" _ _ _ _El _
    FANTASMA _estaba _en _la _habitación _";
    FLASH 1;a; FLASH 0;" _ _ _ _ _"
4040 REM El FANTASMA
4050 REM La boca del FANTASMA
4060 INK 4
4070 PLOT 103,47
4080 DRAW 52,0: DRAW 0,16: DRAW -8, -8:
    DRAW -9,8: DRAW -9, -8: DRAW -9,8:
    DRAW -9, -8: DRAW -8,8: DRAW 0,-16
4090 REM Los ojos del FANTASMA
4100 INK 0
4110 CIRCLE 110,90,7: CIRCLE 146,90,7
4120 REM Perfil del FANTASMA

```



```

4130 PLOT 79,25: DRAW 0,60: DRAW 104,0,
-PI: DRAW 0,-60
4140 GOSUB 65000
4150 PRINT:PRINT:PRINT:PRINT:PRINT:
PRINT:PRINT:PRINT:PRINT:PRINT:PRINT:
PRINT:PRINT:PRINT:PRINT:PRINT
4160 PRINT INK 1;"Tus_puntos_hasta_
ahora_son_"; FLASH 1; BRIGHT 1; Puntos
4170 PRINT INK 1;"Gracias _por _el _
juego,_";a$
4180 GOSUB 60000
4190 PRINT INK 1;"Ahora,_pula _ENTER."
4200 INPUT z$: CLS : RETURN

```

```

5000 REM Un bloque de color
5010 FOR x = a TO b
5020 FOR y = c TO d
5030 PRINT AT x,y; INK e; PAPER f;g$
5040 NEXT y
5050 NEXT x
5060 RETURN

```

```

6000 REM Tonada alegre
6010 BEEP 1,5: BEEP .5,9: BEEP .5,0: BEEP 1,11
6020 BEEP 1,5: BEEP .5,0: BEEP .5,5: BEEP .5,9:
BEEP .5,11: BEEP 1,12
6030 RETURN

```

```

6500 REM Tonada siniestra
6510 BEEP 1,5: BEEP .5,9: BEEP .5,0: BEEP 1,11
6520 BEEP 1,-15: BEEP .5,-13: BEEP .5,-15: BEEP
.5,-17: BEEP .5,-15: BEEP 1,-13
6530 RETURN

```

```

7000 REM Juego nuevo o final
7010 BORDER 5
7020 PRINT:PRINT:PRINT:PRINT:PRINT:
PRINT
7030 PRINT INK 3;"Si _tú _quieres _otro _
juego _de"
7040 PRINT
7050 PRINT INK 2;"-----
FANTASMA"
7060 PRINT


```



```

7070 PRINT INK 3;a$;"_entonces_pon_
y_por_SI"
7080 PRINT
7090 PRINT INK 3;"_o_n_
por_NO."
7100 INPUT b$: CLS: RETURN

```



▲ Cuando hayas procesado el programa varias veces, piensa en cómo podrías mejorarlo.

Procesa el programa unas veces para disfrutar jugando con él. Piensa ahora si puedes mejorarlo. Por ejemplo, considera las siguientes preguntas:

- 1) ¿Puedes aumentar el interés del juego? Quizás quieras añadir líneas que te digan si tu respuesta era mayor o menor que el número correcto, como hiciste para el programa "ADIVINAME". Podrías encajar esto entre las líneas 3030 y 3050, donde por el momento hay cuatro líneas en blanco en la pantalla.
- 2) ¿Puedes mejorar los pequeños detalles del programa? Quizás quieras cambiar alguno de los sonidos, o los colores de la casa.
- 3) ¿Puedes mejorar la estructura general del juego? Por ejemplo, te podría gustar que el dibujo del fantasma apareciera más de una vez. Las líneas 4040 a la 4130 que dibujan el fantasma se podrían tornar en líneas GOSUB empezando en 8000. Luego podrías llamar a las líneas del fantasma entre la 3020 y 3030 y acompañarlas con unos pocos compases de música fantasmal, de forma que estén en pantalla unos pocos segundos. Luego podrías borrar la pantalla y continuar con el mensaje 'mala suerte' de la línea 3030.

Resumen

Ya has aprendido a usar variables de nombres largos por variables numéricas. Aún más importante, has aprendido a usar la instrucción `1 + INT (RND * (un número))`. Esta es una de las instrucciones más útiles para inventar juegos. (¡Podrías volver al juego "ADIVINAME" e intentar convertirlo en un juego para una persona, haciendo que el ordenador escoja al azar un número entre el 1 y el 20!)



Cómo almacenar y cargar programas

Cómo almacenar programas

Has aprendido a poner un título en la primera línea de cada programa, no usando más de 10 letras y/o números, entre dos series de comillas. Por ejemplo, la primera línea del programa de las cartas de agradecimiento era la siguiente:

```
10 REM "GRACIAS"
```

Los espacios entre palabras se *pueden* incluir en el título teniendo en cuenta que juntos unos y otros el total del número de letras, números y espacios no sobrepase de diez. Para almacenar en cinta el programa "GRACIAS" debes:

- 1) Teclear la lista de líneas del programa "GRACIAS" en el ordenador.
- 2) Escribir:

```
SAVE "GRACIAS"
```

Encontrarás la palabra clave SAVE en la tecla S.

3) De la parte de atrás del teclado del ZX Spectrum +, desconectar la toma *EAR*.


4) Pulsar ENTER, y entonces aparecerá el mensaje siguiente:

Start tape, then press any key.

Si has intentado almacenar un título con *más* de diez números, letras y espacios, entonces en lugar del título aparecerá el siguiente mensaje y te obligará a empezar de nuevo:

F Invalid file name, 0:1


5) Pulsar el botón RECORD de tu magnetófono (en algunos también tienes que pulsar PLAY).



6) Pulsar cualquier tecla *excepto* CAPS SHIFT o SYMBOL SHIFT.

7) Observar la pantalla. En el área borde verás rayas rojas y azules anchas, luego rayas azules y amarillas estrechas, después una pantalla en blanco, luego más rayas anchas rojas y azules y luego más rayas estrechas azules y amarillas. Después, las rayas deberán desaparecer y el ordenador imprimirá un mensaje cerca de la parte inferior de la pantalla, diciendo:

Ø OK, Ø:1



▲ Cuando estés verificando o cargando un programa, verás en la pantalla anchas rayas azules y rojas.



▲ Las rayas rojas y azules deberán estar seguidas, a intervalos, por cortas apariciones de rayas azules y amarillas.

Cómo comprobar que el programa ha sido almacenado correctamente

- 1) Vuelve a poner la clavija en la toma *EAR* de la parte de atrás del teclado.
- 2) Rebobina la cinta justo hasta antes de donde almacenaste el programa "GRACIAS".
- 3) Teclea:

VERIFY "GRACIAS"

Encontrarás la Palabra Clave VERIFY en la parte cuadrada de la tecla R.

- 4) Pulsa ENTER y, en seguida, el área borde de la pantalla se quedará en blanco.
- 5) Pulsa el botón PLAY de tu magnetófono. El área borde cambiará entre azul claro y rojo, hasta que el ordenador compruebe que está escuchando el principio del programa "GRACIAS". Entonces verás rayas anchas rojas y azules, luego rayas estrechas azules y amarillas y después aparecerá en la pantalla:

Program: GRACIAS

Entonces habrá un borde rojo, luego más rayas azules y amarillas estrechas. A continuación, las rayas desaparecerán y el ordenador escribirá un mensaje, cerca de la parte inferior de la pantalla, diciendo:

Ø OK, Ø:1

Tu programa ha sido almacenado en cinta y ya puedes desconectar el ordenador sin perderlo.

Cómo cargar programas

Para cargar el programa en el ordenador:

1) Teclea:


LOAD "GRACIAS"

Encontrarás la palabra LOAD sobre la tecla J.

- 2) Pulsa ENTER y el área borde se pondrá blanca.
- 3) Rebobina la cinta hasta justo antes del comienzo del programa "GRACIAS".
- 4) Pulsa el botón PLAY de tu magnetófono. En la pantalla aparecerá lo mismo que cuando pulsaste el botón PLAY al verificar el programa. Cuando el mensaje "OK" aparezca, el programa estará cargado y preparado para ser procesado.

Algunos problemas normales de la verificación o carga de programas

- 1) El magnetófono puede estar puesto con un volumen muy bajo, en cuyo caso deberás subir el volumen.
- 2) El magnetófono puede estar sucio, en cuyo caso deberás poner una cinta limpiadora.
- 3) Las pilas pueden estar gastadas, en cuyo caso deberás instalar nuevas pilas.
- 4) Puede que hayas empezado la cinta en la mitad del programa, en lugar de justo antes del principio. Rebobina la cinta e inténtalo otra vez.
- 5) Puedes haber cometido un error al teclear el nombre del programa que estás tratando de verificar o almacenar. Si esto es posible, inténtalo de nuevo.
- 6) Puedes haber olvidado poner la clavija en la toma EAR. En ese caso, comienza de nuevo. (O quizás olvidaste quitarla cuando almacenaste el programa. Para comprobarlo, desenchufa el magnetófono del ordenador y escucha la parte de la cinta en que supones que estaba almacenado el programa. ¡Si estaba almacenado, oirás algunos ruidos muy fuertes!)



▲ *Si no está bien ajustado el control del volumen del magnetófono, puedes tener problemas al cargar los programas. Puede ser que necesites ajustar el volumen unas cuantas veces antes de encontrar la posición correcta para el magnetófono.*



Una lista de palabras de programación

Aquí encontrarás todas las palabras de programación que has aprendido, junto con un pequeño resumen de cómo se usan y lo que hacen. Presta atención a este capítulo, pues encontrarás algunas nuevas formas de hacer las cosas.

BEEP (parte cuadrada de la Z) Con esta instrucción se crean sonidos.

BORDER (parte redonda de la B) Esta instrucción seguida del número que controla el color cambiará el color de la zona alrededor del borde de la pantalla. Así:

BORDER 1

dará azul oscuro, y así sucesivamente.

BRIGHT (parte cuadrada de la B) Hay dos niveles de brillo en la pantalla. Después de la instrucción:

BRIGHT 1

todo lo que esté escrito en la pantalla será más brillante.

BRIGHT 0

vuelve lo escrito a su brillo normal. Puede usarse **BRIGHT** dentro de una sentencia **PRINT**:

PRINT BRIGHT 1; "Feliz";

BRIGHT 0; ", _ triste."

CIRCLE (parte cuadrada de la H) Esta instrucción dibujará un círculo. A continuación van tres números:

CIRCLE 50,60,25

Los dos primeros números dan las coordenadas del centro del círculo y el tercer número da el radio del círculo.

CLS (parte redonda de la V) Esta instrucción limpia todos los textos y gráficos de la zona central de la pantalla, y la colorea con cualquiera de los colores que se hayan dado. (Si no se ha dado ningún color, aparecerá en blanco).

DRAW (parte redonda de la W) Esta instrucción dibujará una línea recta desde el último punto hasta uno nuevo. Por ejemplo:

DRAW 30,60

le dice al ordenador que dibuje una línea 30 puntos a la derecha y 60 hacia arriba desde la posición actual. Si es el primer punto, el ordenador dibujará desde la esquina inferior izquierda. Esta instrucción también puede dibujar un semicírculo usada con **PI** (sobre la M). Así:

DRAW 30,60,PI

dibujará un semicírculo en sentido contrario a las agujas del reloj desde el último punto donde estaba anteriormente a otro 30 unidades hacia la derecha y 60 hacia arriba. (Escribiendo un signo - (menos) delante de **PI** cambiará el dibujo del semicírculo al sentido de las agujas del reloj).

FLASH (parte cuadrada de la V) Después de la instrucción:

FLASH 1

cada cosa escrita luego en la pantalla 'parpadeará' en cuanto los colores papel y tinta cambian de uno a otro. La instrucción:

FLASH 0

para el centelleo. **FLASH** se puede usar

dentro de una sentencia PRINT. Por ejemplo, puedes tener:

```
PRINT FLASH 1;"AYUDA!"
```

FOR (parte redonda de la F) Es la primera instrucción de un bucle FOR ... NEXT. Hace que el ordenador efectúe una tarea un determinado número de veces. Así:

```
10 FOR a = 1 TO 6
20 PRINT a
30 NEXT a
```

imprimirá (en líneas separadas) los números 1, 2, 3, 4, 5 y 6.

GOSUB (parte redonda de la H) Las instrucciones GOSUB son la forma más útil para lograr que un programa esté ordenado y sea fácil de seguir. Por ejemplo:

```
GOSUB 1000
```

envía al ordenador a la línea 1000 y sucesivas hasta que alcance la instrucción RETURN desde donde vuelve a la línea posterior a la GOSUB 1000. Esto significa que una única GOSUB en el programa principal puede usarse para llamar varias veces el mismo juego de líneas.

GOTO (parte redonda de la G) Hace saltar al ordenador hasta una cierta línea del programa.

IF (parte redonda de la U) Se usa al comienzo de una instrucción IF ... THEN para decir que si (IF) algo es verdad, entonces (THEN) el ordenador debe hacer una cierta cosa. Por ejemplo:

```
IF a = b THEN PRINT "Bien!"
```

INK (parte cuadrada de la X) Se usa con un número para cambiar el color de lo que está escrito o dibujado en la pantalla y también la parte tinta (INK) de los símbolos Gráficos. Así:

```
INK 3
```

significará que las palabras o líneas dibujadas o la parte tinta de los símbolos Gráficos aparecerán en la pantalla en magenta. INK se puede usar dentro de instrucciones PRINT. Así:

```
PRINT INK 3;"Cómo _ estás, _ ";
INK 1;"Jaime?"
```

INPUT (parte redonda de la I) Se usa esta instrucción para introducir información en el ordenador:

1) Con ella puedes introducir números. Por ejemplo:

```
50 PRINT "Mete _ el _ dinero _ que _
tendrás"
60 INPUT b
```

Cuando escribas la cantidad, por ejemplo 15, pulsando luego ENTER, el número 15 será almacenado en un lugar llamado **b**; y si en el futuro das la instrucción:

```
PRINT b
```

aparecerá el número 15 en la pantalla tan pronto como hayas pulsado ENTER.

2) También puedes introducir palabras:

```
210 PRINT "Cómo _ firmarás _
tu _ nombre?"
220 INPUT f$
```

Cuando escribas tu nombre, Luis, por ejemplo, la palabra Luis se almacenará en un lugar llamado **f\$**, y si en el futuro das la instrucción:

```
PRINT f$
```

la palabra Luis aparecerá en la pantalla tan pronto como hayas pulsado ENTER.

INT (parte cuadrada de la R) Ver RND más adelante.

LET (parte redonda de la L) Esta instrucción se usa para almacenar información en el ordenador.

1) Puede almacenar números. Por ejemplo, en la página 19 usaste:

```
LET a = 20
```

Cuando lo hiciste, el número 20 se almacenó en el ordenador con el nombre **a**. Si quisieras añadir 6 a ese almacén, podrías decir:

LET a = 26

o bien, podrías añadir otra línea al programa diciendo:

LET a = a + 6

Podrías también usar la instrucción:

LET Nacimiento = 1969

2) Puede almacenar palabras. Por ejemplo, puedes poner una línea diciendo:

LET a\$ = "reloj"

Esta instrucción introduce la palabra "reloj" en el almacén llamado a\$.

LIST (parte redonda de la K) Una interesante instrucción usada para que el ordenador liste cualquier programa que tenga en su memoria. Si el programa es demasiado largo para salir entero en la pantalla, el mensaje:

scroll?

aparece. Pulsa casi cualquier tecla para continuar; pero CAPS SHIFT y SYMBOL SHIFT no hacen nada y las teclas N y BREAK paran la lista, de forma que puedas añadir líneas o editar las líneas de la pantalla. Si quieres que la lista comience en una línea en particular, pulsa LIST seguida por el número de esa línea.

LOAD (parte redonda de la J) Se usa esta instrucción para cargar un programa en el ordenador.

MODES Han sido explicados completamente en el Capítulo 1.

NEW (parte redonda de la A) Cuando escribes NEW y pulsas ENTER, cualquier línea de programa almacenada en la memoria del ordenador será borrada. También se borrará cualquier cosa de la pantalla.

NEXT (parte redonda de la N) Ver FOR más atrás.

PAPER (parte cuadrada de la C) Esta instrucción, seguida por uno de los números que controlan el color, cambiará el color del fondo detrás de lo que está escrito en la pantalla. Cambiará también el color de la zona papel de los símbolos Gráficos. Así:

PAPER 5

significará que el fondo de las palabras y símbolos y la parte papel de los símbolos Gráficos serán coloreados de cian (azul claro). Para colorear toda la zona central de la pantalla, ver CLS.

PAUSE (parte redonda de la M) La instrucción:

PAUSE 50

para el proceso del programa por un segundo (PAUSE 100 daría dos segundos, etcétera).

PI (parte cuadrada de la M) Ver DRAW arriba.

PLOT (parte redonda de la Q) Esta instrucción pone un punto en la pantalla. Por ejemplo:

PLOT 20,100

sitúa un punto en las coordenadas dadas, en este caso, 20 unidades a la derecha y 100 arriba. Después de una instrucción PLOT, puedes empezar a dibujar desde el punto.

PRINT (parte redonda de la P) Esta instrucción se usa normalmente para imprimir algo en la pantalla. Por ejemplo:

PRINT " "

hace al ordenador imprimir en la pantalla lo que está escrito entre las comillas;

PRINT

por sí sola produce una línea vacía;

PRINT a
escribirá el mismo número previamente almacenado en a;

PRINT a\$
escribirá las letras previamente almacenadas en a\$;

PRINT AT
(AT está sobre la I) escribirá lo que se le diga al ordenador en el cuadro cuyas coordenadas se dan después de las palabras PRINT AT.

REM (parte redonda de la E) El ordenador ignora todo lo que hay en una línea después de REM. Por tanto, las instrucciones REM son muy útiles para recordar de que trata un programa. Por ejemplo, en el programa "FANTASMA" tenías:

1000 REM Introducción

2000 REM Casa fantasma y adivinar etcétera. REM también puede usarse para dar título a un programa, para poderlo almacenar y cargar. Por ejemplo:

10 REM "FANTASMA"

RETURN (parte redonda de la Y) Ver instrucción GOSUB.

RND (parte cuadrada de la T) Hace que el ordenador elija un número al azar (RANDOM).

PRINT RND
escribirá, al azar, un número entre 0 y 1 (pero nunca llega a 1):

PRINT RND*6
escribirá un *número entero* al azar entre 1 y 6.

RUN (parte redonda de la R) Se usa para decir al ordenador que procese el programa que tiene en su memoria.

SAVE (parte redonda de la S) Se usa para almacenar un programa en cinta.

STOP (parte redonda de la A) Esta instrucción señala el lugar donde termina el programa.

THEN (parte redonda de la G). Ver IF.

VERIFY (parte cuadrada de la R) Se usa para comprobar que el programa se ha almacenado en cinta correctamente.



Indice

Varias palabras como PRINT están en muchas páginas. Para que no pierdas el tiempo, este índice sólo da los ejemplos más útiles.

B

BASIC 5, 9
BEEP 14-15, 35-38, 63, 68
Bloques de color 52-57, 61
BORDER 13, 39, 41, 50, 68
BREAK, uso de 6, 14
BRIGHT 42, 68
Bucles 13-14, y ver bucles
FOR ... NEXT

C

CAPS LOCK, uso de 6
CAPS SHIFT, uso de 4-5, 6
CIRCLE 49-68
CLS 29-30, 31, 40, 51, 68
Coloreado de palabras y fondos 39-44
Coma, uso de 22, 23
CONTINUE 14
Cursor 4
Cursor de programa 21-22

D

DELETE, uso de 5
Dibujo de círculos 48-49
Dibujo de líneas 45-46, 62-63
Dibujo de semicírculos 47-48, 63
Dos puntos, uso de 50-51, 60-64
DRAW 10, 45-49, 62-63, 68

E

Edición 21-23, 26
ENTER, uso ordinario 7;
uso como INPUT 31, 33
Escritura de cartas 28-30

F

FLASH 41-42, 60-63, 69
FOR ... NEXT bucle 11,
53-56, 63, 69

G

GOSUB 37-38, 58, 60-64, 69
GOTO 13, 14, 69
Gráficos, ver DRAW,
Dibujos y MODO G

I

IF ... THEN instrucción
27-28, 31-34, 60, 62, 69
INK (tinta) 11, 13, 39-42,
52-56, 60-64, 69
INPUT 20, 24, 29, 31,
60-64, 69-70
INT ver RND

J

JUEGOS
31-34 (ADIVINAME)
58-64 (FANTASMA)

L

LET 13, 19, 56, 58, 70
LIST 10, 11, 70
LOAD y carga de
programas 67, 70

M

Matemáticas 17-18
MODO C (Letras
mayúsculas) 6
MODO E 11-12
MODO G (Gráficos) 12
MODO K 4-5

MODO L 6

MODOS 70, y ver MODOS
C, E, G, K y L

N

NEW 10, 12, 70
NEXT 70 y ver bucles FOR
... NEXT

P

PAPER (papel) 39-41, 51,
52-56, 62-63, 70
PAUSE 13, 70-71
PI 47-48, 63, 71
PLOT 10, 45-46, 62, 71
PRINT 5, 27, 71
PRINT AT 11, 13, 52-56

R

REM 21, 32, 37, 71
RETURN 71
RND 13, 58-59, 60, 71
RUN 9-10, 71

S

SAVE y almacenaje de
programas 65-66, 71
"Scroll" (desenrollado) 26
Solución de problemas,
uso del ordenador para,
17-28
Sonido en programas
35-38, 63 y ver BEEP
STOP 9, 71
SYMBOL SHIFT, uso de 7

T

THEN 71 y ver IF ... THEN
Títulos de programas 21

V

Variables encadenadas
28-33
Variables numéricas
18-28, 58, 60
VERIFY 66-67, 71

GUIAS DE INICIACION

Escritas en estilo sencillo y práctico, las **Guías de iniciación** ofrecen una primera etapa en el uso y aprendizaje de los ordenadores ZX Spectrum, ZX Spectrum + y Commodore 64. Paso a paso, enseñan a escribir programas sencillos, el uso de variables numéricas y encadenadas y la producción de sonidos, colores y gráficos.



Richard Graves, profesor cualificado, y su hijo de 11 años, David Graves, descubrieron que muchos de los libros estándar sobre la introducción a la programación eran demasiado complicados. Juntos han elaborado una guía de programación sencilla que puede ser usada por todas las personas de cualquier edad: la *Guía de iniciación: ZX Spectrum*, ya publicada en esta Colección. La Serie se amplía ahora con la *Guía de iniciación: ZX Spectrum +*.